
Chapter  SEQ Chapter \* ARABIC 1

Turtle Graphics

Using Revolution 

(To the teacher: 

I don’t know which user interface is best. I am inclined toward putting all scripts in the card script and calling them from the message box, using a different card for each chapter. This elimiminates the clutter. Also, be forewarned, this has not been proof read or tested in any way. There will be problems!)


Before we begin to write some programs in Transcript we will spend a little time getting acquainted with the environment. After starting RunRev load the Turtle Graphics document (double click on the icon). This stack has some basic turtle graphics commands built into it.


You should see on the screen two windows. A large window which fills the screen and a small window at the bottom which is called the message box. You can click in the message box and type in any valid Transcript command. (Transcript is the computer language built into Revolution.) For example try: put 2*3 into msg. (The letters msg are an abbreviation for message box.) You should see a 6 appear in the message box. (As a short cut you may type: put 2*3 and a 6 will appear in the message box when you press Return or Enter.) Transcript uses the message box as a default destination for the output if none is specified. If, for some reason, the message box does not appear, you may open it by typing “command/alt m”. Typing command/alt m again will cause it to disappear.


There is another window which is usually invisible but is useful in getting output to the screen. This window is located along the left edge of the screen. To see it in action type the following into the message box: pr 2*3. You should see the answer 6 appear in the upper left corner of the screen. The command pr is an abbreviation for print. To clear this text type ct which is an abbreviation for Clear Text. (By the way, Transcript does not discriminate between upper and lower case. In everything that follows, it makes no difference which you use. For example: Pr, PR, pr, and pR are all the same to Transcript.) You may change this window to a scrolling window by choosing toggle scrolling from the Utilities menu. Choose toggle scrolling again to change back.


Also under the Utilities menu you will find show turtle’s vocabulary. This will take you to a window which the turtle graphics commands and functions. Click on the return button to return to the card you came from.

Turtle Graphics


The “Turtle” in Turtle Graphics is a metaphor to represent a drawing tool. The Turtle has four attributes: an x coordinate, a y coordinate, a heading, and a pen state (up or down). The turtle may be asked to move to a new x-y point. He may be asked to move forward a given distance in the direction of his current heading. The heading of the turtle is measured in degrees, in a counter-clockwise sense from the x-axis. East is a zero heading, North a 90 degree heading and so on. The turtle may or may not leave a line in his wake depending on whether his drawing pen is up or down. 


The advantages of turtle graphics are many. For example, suppose you wanted to draw a pentagon. Using coordinate geometry, it would be necessary to determine the coordinates of the five vertices of the pentagon and draw lines connecting these points—not an easy task. Using local turtle geometry we simply instruct the turtle to move forward a distance equal to the length of one side and turn right through and angle of 360/5 = 72 degrees. If this process is repeated 5 times you have pentagon. This is a very simple algorithm. (An algorithm is a rule or procedure for solving a problem.)


Of course, the turtle is also capable of executing the commands of coordinate geometry. For example he can be instructed to move to the point at which x = 30 and y = 40. 


In the material that follows we will be using both. At times, the coordinate geometry will be simplest and at others, the local turtle geometry.

A Few Simple Commands

1. Commands: st, clean, home, setHome, forward, back, right, left, pu, pd and label. 

st:  
To use the turtle graphics commands it is necessary to initialize the turtle. This is done with the st command (an abbreviation for start turtle). Before beginning any turtle graphics it will be necessary to use this command. (You will notice on the graphics pallet in the “Tools” menu that the pencil becomes the selected tool when you run st. If for some reason this tool becomes deselected, it will be necessary to initialize the turtle again with st. Sometimes it may be possible to simply select the pencil again by clicking on it. The surest method however is to restart the turtle.) This command also places the center of the coordinate system at the center of the screen. 

showTurtle
The turtle is represented by a short arrow. To see the turtle you must execute the showTurtle command. There is no abbreviation. In complicated problems it may help to hide the turtle as he moves more  quickly when he is invisible. 

hideTurtle
The opposite of showTurtle.

clean: 
You may wipe the screen clean with this command. The st command cleans the screen as well, but it also returns the turtle to the home position.

home: 
This command sends the turtle to home base; the origin of the coordinate system. This is usually at the center of the screen but since the home position can be changed (see setHome below) it will send the turtle to wherever the current home base is located. Home not only zeros the turtles coordinates but it also zeros his heading as well.

setHome:
This command is used to relocate the origin of the coordinate system. For example: setHome 30,40 would move the origin 30 pixels to the right and 40 pixels up. These distances are measured from the center of the screen.

forward: 
This command does what you might expect. It moves the turtle forward at his current heading. For example: forward 100, moves the turtle forward 100 screen units (pixels).

back: 
The opposite of forward.

right: 
The command right 30, will turn the turtle clockwise through an angle of 30 degrees.

left: 
The opposite of right.

pu: 
This is an abbreviation for pen up. It lifts the pen so that the turtle leaves no line in his wake.

pd: 
An abbreviation for pen down. It puts the pen down so that the turtle will leave a line behind him as he moves.

Label “x”  
When called, this command types the quoted input at the current turtle position. This command is useful in labeling a figure and sometimes for just locating the turtle.

The following abbreviations will speed up your typing:



COMMAND
ABBREVIATION



forward
fd


back
bk


right
rt


left
lt

Now let’s see how these may be put to work. Type the following commands in sequence in the message box and observer the turtles progress (you won’t see the turtle, but his wake is visible):

startTurtle
(or st)

showTurtle

forward 50
(or fd 50)

right 90
(or rt 90)

back 100
(or bk 100)

left 45
(or lt 45)

forward 100
(or fd 100)

penUp
(or pu)

forward 50
(or fd 50)

label “x”

penDown
(or pd)

forward 50
(or fd 50)

Project 1.  SEQ Project_1. \* ARABIC 1 Use fd, and lt to draw a square 100 units on a side.

Command: 
setxy (short for: set x and y coordinates) and incxy (short for: increment the x and y coordinates).



The command setxy takes two parameters: an x coordinate and a y coordinate separated by a comma. The turtle is moved to these new coordinates (leaving a trail behind him if the pen is down). For example setxy 30,40 will place the turtle at the coordinates 30,40. (We will see later that setxy also accepts a two parameter list.)



The command incxy also takes two parameters: an x increment and a y increment separated by a comma.  The command incxy 5,8 will increase the x coordinate by 5 units and the y coordinate by 8 units. The reference in setxy is absolute while the reference in incxy is relative. The first sets the absolute values of x and y while incxy increases the coordinates relative to the current position of the turtle. 

Project 1.  SEQ Project_1. \* ARABIC 2 Use incxy to draw a square 50 units on a side.
Command:  seth (short for set heading).


The command seth is used to set the heading of the turtle. Angles are measured counterclockwise from the x-axis. For example: seth 30, will point the turtle in a direction 30 degrees above the x-axis. 

Project 1.  SEQ Project_1. \* ARABIC 3 Draw a square 50 units on a side using seth and fd.

Project 1.  SEQ Project_1. \* ARABIC 4 Using seth, fd (or bk), and home, verify that the side opposite the 30 degree angle in a right triangle is one half the hypotenuse.
Chapter  SEQ Chapter \* ARABIC 2

Functions
Introduction


Thus far we have been giving commands to the turtle. We have told the turtle to do something (go forward 100 units, turn left through 90 degrees, etc.). But we can also ask the turtle to speak to us. He is not a great conversationalist.  All he knows is where he’s at and where he’s pointed. We may ask the turtle to tell us his heading, or his x and y coordinates, or his distance from some point on the screen. 


Unlike commands such as FORWARD or RIGHT, the turtle reports his position or heading through functions. Functions differ from commands in that they return a value. A command on the other hand performs a task. Functions have a different form from commands. They are always called with trailing parentheses. The turtle functions for example are: xyCor(), xcor(), ycor(), heading(), distance(), and direction(). Whenever Transcript returns a value it has been called by a function and that function ends in parentheses. These functions differ from those we are accustomed to in mathematics. In mathematics we are accustomed to functions written in the form: f(x) = x2; this equation reads: f of x equals x squared. The variable x is called the argument of the function f. There is always an argument to the function f. We never see f(). Whenever we have a turtle graphics function such as xyCor(), or xcor(), these should be read as follows: xyCor of the turtle, or xcor of the turtle, or heading of the turtle. Within the empty brackets the argument of a turtle graphics function the words tof the turtlel are always understood—well, almost always. Later we will need a second animal, the monkey. At that time xcor() will stand for: xcor of the monkey. 


There are many functions built into Transcript. For example sin(pi), tan(pi/4), random(10), round(3.14159), seconds(), sqrt(2), mouseLoc(), etc. Some of these functions have numbers within the parentheses and some do not. It depends on whether Transcript needs some input before it can return an output. For example, before Transcript can return a value for the square root, it must be given a number of which to take the square root. On the other hand, it doesn’t need any input to return the position of the mouse or the number of seconds since some fixed date. There is however an implied argument to every function. For example, the implied argument in seconds() is eof the clock.
 The implied argument in mouseLoc() is Tof the mouse.h The important point to remember is that if a procedure returns a value, it is a function and must end in parentheses. If it simply performs a task it is a command and no parentheses are needed. One of the most common mistakes in programming in Transcript is forgetting these parentheses in functions.

1. Functions: xcor(), ycor() , and xyCor(). 


The function xcor() and ycor() return the x and y coordinates of the turtle respectively. To see these functions in action run the following commands:


st


fd 100

lt 90

fd 50

put xcor()


put ycor()
You will see the x and y coordinates printed in the message box. (You cannot use pr xcor(0). This is a Transcript bug.)


The function xycor() is a little different. Type into the message box: put xycor(). You should see: 100,50—the x and y coordinates of the turtle separated by a comma. This is a special kind of Transcript container called a list. A list is: A sequence of words or numbers separated by commas. As another example of a list we might have:


this, is, a, list of, 10, words and, 2, numbers

We refer to specific elements of the list by their item number. For example the first item in the above list is the word “this”. The forth item is the pair of words “list of”. The seventh item in the list is the number “2”. 


The function xycor() returns a list of two numbers—the x and y coordinates. We shall discuss lists in more detail later. 

Project 2.  SEQ Project_2. \* ARABIC 1 Determine the square root of 2 using put, and xcor(). (Hint: What is the hypotenuse of a right triangle whose legs are both equal to the square root of 2?)

2. Function: heading()

The function heading() returns the direction in which the turtle is pointing.

Type the following into the message box:


setxy 50,100


seth 45


fd 100


put heading()


put xyCor()
You should see:


45


120.710678,170.710678

The first number is the heading of the turtle and the pair of numbers: “120.710678, 170.710678” is a list of two items. The first item is the x coordinate and the second item is the y coordinate of the turtle. 

3. Functions: direction(?,?) and distance(?,?).


The two functions direction and distance need some input before they can return an output. The function direction(?,?) takes two inputs: an x coordinate and a y coordinate. They are separated by a comma and enclosed in the parentheses. (The input is not a list even though it is a pair of numbers separated by a comma. There is a very confusing usage of commas in Transcript and we shall return to this point latter. For the moment, just remember that direction takes two inputs separated by a comma.) The effect of direction is to return the angle at which the turtle should look in order to see the given coordinate. It does not rotate the turtle to this angle. For example try the following in the message box:


st


seth direction(100,100)


put heading()


seth direction (sqrt(3), 1)


put heading()    

put  direction(100,100)

You should see 45,30, and 45 printed in the message box. The reason for this is that the turtle has been placed at the origin with the command st (start turtle). He was then directed to set his heading toward the xyCor (100,100). This point is located at an angle of 45 degrees from his current position. Therefore when asked to print his heading he returned 45 degrees. He was then directed to set his heading toward the xyCor (sqrt(3),1). This point is at an angle of 30 degrees, and so he tells us his new heading is 30 degrees. He is then asked to print the direction of the xyCor (100,100). This is 45 degrees even though his heading is still 30 degrees.


Notice that direction(100,100) is a function which returns a number. The command seth acts on this output by setting the heading to this number. Notice as well that heading() is also a function which returns 45 to the command pr. We see here a general rule:


General rule: All functions must be preceded by a command. The command acts on the output from the  function, for  example,  seth might  takes as its input direction(100,100). Without the command the output of the function would be left hanging. (As a rule Transcript will return the results of a function to the default position, the message box, unless some command tells it what to do.) 


It should be mentioned that choosing to adopt a notation which distinguishes between functions and commands is not a logical necessity. There are some computer languages which use the same notation for commands and functions—a custom which I prefer. That Transcript should make the distinction is a matter of choice—a decision made by the authors of Transcript. 

Project 2.  SEQ Project_2. \* ARABIC 2 Move the turtle from home to the point (200,100) using seth, distance(), direction(), and fd. Verify that the turtle has moved to this point by running put xycor().

Chapter  SEQ Chapter \* ARABIC 3

User Defined Procedures

Introduction


Thus far we have been running Transcript commands in the immediate mode; running one line at a time in the message window. Now we will use Transcript as a programming language for computation and graphics representation.


Let us return to our project for drawing squares. We would like to automate this process by writing a "sentence"; a string of words put together to form a single grammatical unit. When we give this sentence a name we have a "procedure". Once a procedure has been defined it assumes, for all practical purposes, the same status as the built-in Transcript primitives. The language has, in effect, been extended to include our new user defined procedure. For this reason, Transcript is said to be an "extensible" language. 


To define a procedure, we must enter the editor. In Revolution there is a separate editor for each card. (To go from card to card use the Go menu or the arrow keys. You may generate a new card by selecting the item “new card” under the “Object” menu.) The card editor is entered by typing command-shift-C. (Each object in Revolution has its own set of procedures. There is a different editor for each card, each button, each field, each background, and the stack itself. There is a hierarchy which determines how messages are passed through the system. Understanding this hierarchy is vital to programming Revolution. Since our objective is using Transcript as a tool to learn some mathematics and physics we will take up only those features of Revolution which are necessary for our purposes.) 


 Type command-shift-C (or control-shift-C) to enter the editor of the current card. Type the following procedure into the edit window:

on square
--Define the procedure named: square. 

 st

--Initiate the turtle (Start Turtle). 

 fd 50

--Go forward 50 pixels.

 lt 90

--Turn left.

 fd 50

--Repeat the process four more times.

 lt 90

 fd 50

 lt 90

 fd 50

end square
--End of definition.

There are many things we should notice about this procedure. All commands begin with “on” followed by the name of the command. (We will see later than all function definitions begin with the word function.) The command ends with “end” followed again by the name of the command. The word “on” translates into: “Whenever I (Revolution) encounter the word square I will execute the sequence of statements between on square and end square”. There is only one statement per line, and the body of the procedure is indented. This indentation (formatting) is executed by the editor every time the return or the tab key is pressed. In complicated procedures, the formatting is very helpful in understanding the grouping of ideas. 


Sometimes it is helpful to explain the reasoning behind a line of programming code. To do this a programmer inserts a comment. All comments begin with two successive minus signs: “--”. These are not read or executed by Revolution. They exists solely for the benefit of the reader. You may omit them from your scripts if you wish. Generally speaking they  are considered good programming practice.


To enter the procedure into memory press the enter key; to leave the editor press the enter key a second time. The editor will disappear and you will find yourself before the card window once again. (If, for some reason, you do not wish to record the procedure(s) press the revert button and the original procedure will be restored.) Type square in the message box and see the script you wrote carried out by Revolution.

Iteration


The above definition of square is unnecessarily cumbersome. It may be made simpler by using the Transcript iteration commands—namely the repeat structures. We may streamline the definition of square as follows:

on square

 st              
--Start the turtle.

 repeat 4 times
--Draw the square.

   fd 50           

   lt 90

 end repeat

end square

In this use of repeat, the commands fd 50 and lt 90 are repeat four times. The instructions begin with repeat and terminated with end repeat. (The word times is optional.) 

Other forms of the repeat structure are:



repeat forever



repeat until {some condition is satisfied}



repeat while {some condition is satisfied}



repeat with {counter} = {starting number} to {ending number}


With repeat forever the instructions are repeated forever. If you can’t wait that long, type command period to stop the iteration. (Repeat forever has the same effect as repeat.)


With repeat until the instructions are repeated until the condition is satisfied.


With repeat while the instructions are repeated as long as the condition is satisfied.


With repeat with the instructions are repeated with the counter variable increasing by one with every iteration. 


(There is also a repeat for each… but this  is beyond the scope of this program.)


For example, to print out the “nines” in the times table:

on nines

 repeat with i = 1 to 9

  pr i*9

end repeat

end nines

--Do the same thing another way.

 put 1 into i

 repeat until i = 10

  pr i*9

  add 1 to i

 end repeat

--Yet another way.

 put 1 into i

 repeat while i < 10

  pr i*9

  add 1 to i

 end repeat

The “nines” portions of the times table will be printed three times—once by each of the three repeat structures. (We will define the command put in the next section.)


It is possible to leave a repeat structure at any time with the command: exit repeat. The procedure will resume with the first statement after end repeat. So that we may rewrite the nines procedure:

on nines

 put 1 into i

 repeat forever

  pr i*9

  if i > 9 then exit repeat

  add 1 to i

 end repeat

end nines


It is also possible to cancel one iteration of a repeat structure with the command: next repeat. (We will define the conditional if—then in the next section.) For example a less articulate form of the “nines” table is:

on nines

repeat with i = 0 to 9

 if i = 0 then next repeat

 pr i*9

end repeat

end nines

This form will skip the term for which i = 0 and so the product 0*9 will not be executed. The procedure will continue with the remaining nine steps.


It is also possible to use a repeat structure within a repeat structure. For example we may print out the entire times table with the procedure:

on timesTable

 repeat with i = 1 to 9     

  repeat with j = 1 to 9

   pr i*j                --Print out the entire times table.

  end repeat

 end repeat

end timesTable

Notice how the formatting (indenting) nests the statements to be repeated for easier reading. You won’t be able to see the entire table in the card field using the “pr” command. Much of it will disappear off the bottom of the screen. One way to see the entire table on the screen without scrolling is to rewrite timesTable:

on timesTable

 repeat with i = 1 to 9

  repeat with j = 1 to 9

   put i*j into item j of outPut

  end repeat

  pr outPut

 end repeat

end timesTable

This procedure will place commas between each term and print the one-times, the two-times, etc. as individual lines. (We will define item in the next section.) Another way to scroll through the times table is to choose Toggle Scrolling from the Utilities menu.

 Project 3.  SEQ Project_3. \* ARABIC 1 Write a procedure spinSquare which rotates a square 10 times through an angle of 36°, i.e.



on spinSquare



st



repeat 10 times



...                 --Put your code here. Use the SQUARE procedure above.



end repeat



end

This procedure should use your previously defined SQUARE procedure just as if it were a built-in procedure. 

This is what spinSquare should produce.


The project above the illustrates sentence-building nature of Transcript. A task is executed in Transcript by building a "vocabulary" (e.g. SQUARE, CIRCLE, BOX, ORBIT, etc.). These words may be strung together in what is often called a "super procedure". The larger problem is thereby broken down into many smaller units.

Project 3. SEQ Project_3. \* ARABIC 2
Use repeat to draw a pentagon 100 units on a side. 

Project 3. SEQ Project_3. \* ARABIC 3
Use repeat to draw a five pointed star 100 units on a side. To find the proper angle, try walking your way around the star and see what angle you turn through in making the complete round trip.

                 A pentagon                            A five pointed star 

Project 3. SEQ Project_3. \* ARABIC 4
Using repeat and seth, draw a stair with 10 steps. The width of the tread is 20 units and the height of the riser is 9 units.

Project 3.  SEQ Project_3. \* ARABIC 5
Using the project above, simulate a boat crossing a stream. The velocity of the stream is 3 ft/sec at a heading of 0 degrees, and the velocity of the boat is 4 ft/sec at a heading of 90 degrees. The stream is 80 ft wide. (Let the stream carry the boat downstream for 1 second. Stop the stream and row the boat cross-stream for 1 second. Repeat the process as often as necessary to complete the crossing of 80 ft. This project should bear a great similarity to project 12.)

There are three new Transcript words used in the above scripts which we haven’t yet defined. It is a tribute to the naturalness of the language that you probably understood their meaning without formal definitions. However we must now define these additions. The three new words are: put, add, and if-then.

Commands: PUT, ADD, SUBTRACT, MULTIPLY, and DIVIDE


In Transcript, information may be stored temporarily in variables. To put a value (number or word or sentence or whatever) into a variable we use the put command. For example to put 3.14159 into the variable x:


put 3.14159 into x

The variable x now takes on the value 3.14159. (By the way, the number pi is included in Transcript. Try typing: pr pi.) We may now use the variable x as if it were the number itself. However, unlike the number, the variable may be changed—hence the name, variable. We may change x in many ways. For example:


add 3 to x


put x + 3 into x


subtract 4 from x


put x - 4 into x


multiply x by 5


put 5*x into x


divide x by 6


put x/6 into x

All of these operations change the value of x. The first operation increases x by 3. The second operation does exactly the same thing. Take your pick. Each of the remaining three pairs of operations are equivalent as well.

Project 3.  SEQ Project_3. \* ARABIC 6
Frank Meyers, in the town of Mayberry, is about to be evicted from his home by Sherif Andy Taylor when he discovers a 100 year old municipal bond in the amount of $100. It pays eight and one half percent interest compounded annually. How much is the bond currently worth?

Project 3.  SEQ Project_3. \* ARABIC 7
Zeno was a Greek philosopher who proposed the following paradox: Suppose you wanted to travel from Athens to Sparta and the distance between them were 1 mile. (I’m simplifying the numbers.) Traveling at one mile per hour how long would it take to reach your Sparta? Zeno said that it would take an infinite time and here is his reasoning: He divided the distance he had to travel in half. It would take him 1/2 an hour to travel this 1/2 mile. He then divided the remaining distance in half so that it would take him 1/4 hour to travel this 1/4 mile. He continued this process indefinitely, always having one half the remaining distance to travel. He claimed that the sum of this infinite sequence would itself be infinite. Show that he was wrong. Hint: Use repeat, add, and divide. Add the time it take him to make say the first 30 legs of his journey. Remember that time = distance/velocity.


You can actually derive a value for the sum of this infinite seriers. Hint: Multiply the infinite series (1/2 + 1/4 + 1/8 + ......) by 2.

Conditional: IF ...THEN...ELSE


There are many instances when we wish to have an action taken depending on the outcome of some condition. For example, to bounce the turtle around a room 100 ft on a side:

on bounce

 st

 put 5 into vx

 put 4 into vy

 repeat 4        

--Draw the room.

  fd 100

  lt 90

 end repeat

 repeat forever

  incXY vx,vy

  if xcor() > 100 or xcor() < 0 then    
--If out of bounds along the x axis then bounce.

   multiply vx by -1

  end if

  if ycor() > 100 or ycor() < 0 then   
 --If out of bounds along the y axis then bounce.

   multiply vy by -1

  end if

 end repeat

end bounce

Whenever the turtle gets outside the room, he is bounced back. (Use command period to stop the program.) The general conditional form we shall use is:

if {condition} and/or {condition} and/or {condition} ... then

  {do something}

else

  {do something else}

end if

For example, if we wanted to write a square root procedure (the built-in Transcript function for square root is sqrt()) which checks for negative numbers we would define:

on squareRoot x

 if x < 0 then

  pr "Imaginary root."   --The message is returned and the function stops.

 else

  pr sqrt(x)

 end if

end squareRoot

If we type squareRoot(-4) we obtain the message: Imaginary root.

If the {do something} is short (just one statement) it is possible to use the shorter form:



if {condition} then {do one thing}

In this case it is not necessary to terminate the conditional with end if. For example in the bounce procedure we could have said:

if xcor() > 100 or xcor() < 0 then multiply vx by -1

Whether you use the long form or the short form depends on how many statements need to be executed. If it is only one, you may use the short form which does not use end if.  If it is more than one, you must use the long form and terminate the condition with end if.

Project 3.  SEQ Project_3. \* ARABIC 8
Modify the bounce procedure above to allow the ball to escape from a hole 25 units long as shown in the figure below. (One way to do this is to use two repeat structures. The first is: repeat until {condition}, where the {condition} is satisfied only when the ball passes through the hole. The second repeat structure applies after the ball escapes and is of the form: repeat forever.)
A ball bounces within a box 100 units on a side.

There is a hole 25 units wide in the left side of 

the box from which the ball eventually escapes.

Lists

One of the most useful containers in Transcript is the list. We have mentioned lists earlier. A list is a series of words or numbers bounded on each end by quotation marks and separated by commas. Type the following in the message box:


put  “The, girl, had, 2, big dogs, and, 3, cats” into listVariable

The variable listVariable is bounded on both ends by quotes and contains 8 items separated by commas. To obtain access to these items we use the item command. For example, type into the message box:


pr listVariable


pr item 2 of listVariable


pr item 5 of listVariable


pr (item 4 of listVariable) * (item 7 of listVariable)

You should see:


The, girl, had, 2, big dogs, and, 3, cats


 girl


big dogs


6

Project 3.  SEQ Project_3. \* ARABIC 9 We are going to study a little genetics. In particular we will see why it is that the incest taboo is so universal among all peoples. Your task will be to demonstrate how dangerous it is. 


At every site on a person’s chromosomes there two genes (actually two alleles). A child will receive one of these genes from its father and one from its mother. Suppose we say that the father has genes a and b at a given site and the mother has genes c and d. We might represent this in a script in the following way:


put "a,b" into father


put "c,d" into mother 

Suppose these parents have two children, “brother” and “sister”. At the corresponding site they will receive randomly one gene from the father and one from the mother. For example the brother might wind up with “b,c” and the daughter with “a,c”. Since which gene is transferred is random, we need Transcript’s random function. The function random(10) will return an integer between 1 and 10. The function random(2) will return either a 1 or a 2, etc. So we might say for the inheritance of the brother:


put item random(2) of father into item 1 of brother

   
put item random(2) of mother into item 2 of brother

We would do the same for the sister. 


Now suppose the brother and sister have an incestuous relationship, that is, have a child. The child will receive one gene, randomly, from the brother and one from the sister:


put item random(2) of brother into item 1 of child


put item random(2) of sister into item 2 of child

Now here’s where the danger lies. The average individual has about four lethal, recessive genes. A gene is recessive if it has no effect on the individual if it is coupled with any other dominant gene. Suppose “a” is a lethal recessive gene and b,c, and d are all healthy dominant genes. If the child should inherit anything other than “a,a” from the brother and sister it will be fine. However if it should inherit “a,a” it will not survive. 


Normally this recessive lethal gene is a very rare. Therefore the chances of two partners having this same gene is very rare. Rare except when the partners have common ancestors, such as the brother and sister above. They both might have inherited “a” from the father and passed it on to the child. We might test for the child’s health as follows:


if child = "a,a" then add 1 to numberOfDeaths

See if you can determine the odds of the child surviving. As a start try the following script:

on taboo

  put 0 into numberOfDeaths

  put "a,b" into father

  put "c,d" into mother

  put 100 into thisManyChildren

  repeat thisManyChildren times

   put item random(2) of father into item 1 of son

   put item random(2) of mother into item 2 of son

   put item random(2) of father into item 1 of daughter

   (etc.)

  end repeat

  pr "Probability that the child will die = " & numberOfDeaths /thisManyChildren

 end repeat


end taboo

Suppose “c” in the mother were also a lethal recessive gene. What then is the likelihood of survival for the child?

Variable Scope


Different computer languages use different rules to determine the scope of a variable.  By “scope of a variable” we mean the range over which the variable has meaning. In Transcript all variables are local unless otherwise defined. (Now what in the world does that mean?) Consider the following two procedures:

on firstProcedure

put 3.14 into x

pr x

secondProcedure    -- This is where firstProcedure calls secondProcedure.

end firstProcedure

on secondProcedure

add 1 to x         -- The secondProcedure doesn’t know what x is.

pr x

end secondProcedure

If these two procedures are run you will get an error message:


Unexpected number here.

The first procedure calls the second procedure which tries to add 1 to x but the second procedure doesn’t know what x is. It has no knowledge of the variable x which appears in the first procedure. Each of these procedures has its own local variables and does not share them with other procedures unless asked to do so. One way to let procedures share variables is to declare them global. 

firstProcedure

global x

put 3.14 into x

pr x

secondProcedure 

end firstProcedure

secondProcedure

global x

add 1 to x

pr x

end secondProcedure

If you now run these procedures, the first will print 3.14 and the second 4.14. The second procedure now shares the value of x with the first procedure. To achieve this sharing, you must declare the variables global in each procedure which uses the variable. The term global is somewhat misleading; share would be better. When you say that a variable is global in a procedure you allow that procedure to share the given variable with other procedures which also declare the same variable as global. 


If there is more than one global variable, they are appended to global and separated by commas. For example:


global x,y,z

will make x, y, and z global variables.


Caution: Generally speaking, it is not a good idea to use a lot of global variables. Too often you will forget what is local and what is not and debugging a set of procedures can be a nightmare. There is another way in which procedures can share information, and that is by passing the information in the form of parameters. We will discuss this topic next.


What we have  just described is the convention in most x-Talk languages. Fortuantely Transcript (as well as MetaCard) has a much better way of allowing procedures to share variables. The collection of procedures for any object (card, button, field,...) is called the script for that object. (What we have called procedures, are called handlers in Transcript. The collection of handlers makes up the script.) It is possible to allow all procedures within  a  script of  a given object (card, button, field,...) to share a variable by delareing it to be local to that object. Thus  at the beginnning of a card script if you declare:

Local x,y,z

Then all procedures that follow within the script of that card script may be shared with each other. (However if you make a similar declaation in a button on that card, the varaibles x,y and z canot be shared. If you  wish other objects, other cards, fields, buttons to share some variable you must  decare it global. For example:

Global pi

Would  allow all objects whaich also declare pi to be global to share its value. The ability to make this declaration for the object is a great convenience. It is cumbersome to have to make such declarations within each separate procedure. 

Parameters


Our programming thus far has been rather rigid. We have written a procedure which will draw a square 50 units on a side. To draw a square 100 units on a side we would need a new procedure. What we would prefer is a single procedure which might be used to draw a square of any size. 


This is accomplished in much the same way as in natural language (English for example). We do not have separate words for a 50 unit square and a 100 unit square. We begin with a generic word "square" and append modifiers.  For example in the sentence: "The big, brown, dog bit the tall, angry, mailman." "Big", "brown", "tall", and "angry" are modifiers (adjectives in this case) which allow us to describe a particular type of dog or mailman and permit us a great economy of language. We don't want to have to create a new word for a "big, brown, dog", or a "tall,angry, mailman". 


In the same way we might want to draw a square with a specific size, in a specific position, and turned at a specific angle. To do so we must be able to add modifiers to SQUARE.


Let us begin with a single modifier for square— a size modifier.  Write the following procedure in the card editor:

on square L

repeat 4 times

 fd L 

 rt 90

end repeat

end square

Run square 50. You should see a square 50 units on a side exactly as before. The command square 50 looks for a procedure named square followed by a modifier. It assigns the value 50 to the modifier L. The modifier L is a local variable which will have the value 50 throughout the rest of the square procedure. If you run square 80 the modifier L is assigned the value 80.


To understand the significance of square 50 it might be compared with the more familiar forward 50. forward is a command which takes one input. The result of the command is to move the turtle forward by an amount equal to the input parameter. In the same way square is a command which requires one input. The result of the command is to draw a square whose size is determined by the value of that input. When we type SQUARE 50 the variable L is assigned the value 50. Try square 75.

Project 3.  SEQ Project_3. \* ARABIC 10 Write a procedure square with three variable modifiers, three input variables. In the editor write a procedure square L, X, Y which will draw a square of the specified length with the lower left corner of the square located at the coordinates (X,Y). Use PU, PD, SETXY, etc. to fill in the blanks in the following:

on square L, X,Y

...

...

end

Project 3.  SEQ Project_3. \* ARABIC 11 
A number is a mathematical quantity with two properties: a magnitude and a sign. A vector is a mathematical quantity with also has two properties: a magnitude and a direction. Many physical quantities are represented by vectors, for example: displacement, velocity, force, electric field, etc. Write a procedure vectorFd which takes a single list as an input variable. The first item in the list is a distance, and the second item is a direction. The effect of vectorFd is to set the turtle in the given direction and then move him forward the given distance. Fill in the blanks:

on vectorFd vector

seth ....

fd ...

end vectorFd

Type vectorFd “100,45” and see the turtle move forward 100 units at an angle of 45 degrees.

Project 3.  SEQ Project_3. \* ARABIC 12 Addition of vectors. The rule for addition of vectors is the same as the rule for adding displacements. We add displacements by simply allowing one displacement to follow the other and the sum is just the net displacement from the beginning of the first to the end of the last. For example the sum (or resultant) the the three vectors A, B, and C is the vector R. 

Determine the vector sum of two vectors. The first is 50 units at 30 degrees and the second 100 units at 60 degrees. Determine the magnitude and direction of the sum. 


 Another word for these input modifiers is parameters. We say that the procedure square takes three parameters, a length parameter, a x position parameter, and a y position parameter. Notice that the input parameters are separated by commas. These three inputs are not a list of three items. If a procedure takes one or more lists as parameter inputs, the lists must be enclosed in quotes. For example, suppose we wanted to write a procedure to draw a line between two points (we will do this later). The procedure might begin as follows:


on lineBetween P,Q


{do the right stuff to draw a line between points P and Q}


end lineBetween

where P and Q are lists representing the x and y coordinates of two points. Suppose P has coordinates 30,40 and Q has coordinates 50,70. Then to draw a line between points P and Q we would type: 


lineBetween “30,40”, “50,70”

There are three commas here—the two that separate the items in the two lists and the single comma that separates the two parameters that are the input to lineBetween. This is the potential for confusion that exists with the usage of commas in Transcript that we mentioned earlier. It is unfortunate that Transcript uses the same delimiter, the comma, in two different contexts. 


In order to simplify the turtle graphics language we have allowed either two parameters or a single list as the input to some commands and functions. For example you may use either setXY 30,40 or setXY “30,40” and the result is the same. The command setXY looks to see in there is just one input—a list. If there is, it picks the list apart and puts the first item into the first parameter and the second item into the second parameter and then goes about its business. The same is true for direction and distance. This is all a bit clumsy but we felt it was better than the alternative which is to write two sets of procedures: one for the two parameter input and one for the single list input.

Project 3.  SEQ Project_3. \* ARABIC 13
 Write a procedure which draws a line between two points P and Q. If you run: lineBetween  “30,40” , “50,70”, a line should be drawn between the two points. 

Project 3.  SEQ Project_3. \* ARABIC 14 
Write a procedure triangle which draws a triangle connecting the points P, Q, and R which are “200,40”, “80,180”, and “250,170” respectively.  Use the command lineBetween defined above.


We pointed out at the end of the last section that input parameters are a way of passing information to procedures without declaring global variables. For example, you will later develop a procedure for drawing the orbit of planets revolving about the sun. Let us call the procedure orbit, and let the initial x and y components of the velocity be vx and vy.  We might write a procedure:

on setUp 

 st
--Start the turtle.

 put “150,0” into startingPoint
--Set up his starting point.

 setXY startingPoint
--Place the turtle.

 orbit 5,10                  --Call orbit and pass the values 5 and 10 to the x and y velocities.

end setUp

on orbit vx, vy
--Orbit begins with vx = 5 and vy = 10

 repeat forever

  {move the planet using the velocity components vx and vy}

  {find the acceleration and determine the new velocity}

 end repeat

end orbit

The procedure setUp puts the planet in the proper starting position and then calls orbit with initial x and y components of velocity of 5 and 10 respectively.  These velocity components are passed to orbit through the two parameters vx and vy. Furthermore, we might determine the acceleration by calling some other procedure passing the x and y coordinates to that procedure through two new parameters. In short, parameters are a very useful method of passing information back and forth between procedures without declaring a lot of global variables. 


You should also be aware that the scope of a procedure’s parameters is always local. They cannot be made global. They go out of existence as soon as the procedure using them stops running. If you were to run the following procedure:


on test num


  global num


  pr num


end test 

Transcript would respond with the message:


Already have a local variable named num.


Not only would we like to pass information to other procedures but we would like to get information back from other procedures. This is done through the use of user-defined functions which we will consider presently. But first, let’s consider another application.

The Geometry of the Circle


Over two thousand years ago the Greeks discovered what is probably the most recognized theorem in geometry. We are going to rediscover this theorem using the turtle. 


The theorem tells us something about the relationship between the circumference of a circle and its diameter. The property of the circle we would like to investigate is the ratio of the circumference to the diameter, that is

Project 3.  SEQ Project_3. \* ARABIC 15 First write a procedure to draw a circle. Fill in the gaps in the following procedure:


on circle step


st


seth 90


put 36 into number


repeat number times


  fd step


  {etc.}


end repeat


end circle

where step is the length of the steps taken by the turtle in walking around the circumference of the circle and num is the number of steps the turtle takes to complete the trip. (If step is 2 what is the circumference of the circle?)

Project 3.  SEQ Project_3. \* ARABIC 16
 The trouble with the above procedure is that it doesn't allow us to measure the diameter. Suppose, however, that we taught the turtle to draw a semicircle. When the turtle completes one half of the circle we should be able to determine the diameter from his xcor(). Calculate the ratio of the circumference to the diameter by completing the following procedure:


on circle step


st


seth 90


put 36 into num


repeat num/2 times


  {Draw the semicircle}


end repeat


put ... into circumference


put ... into diameter


pr circumference / diameter


end circle 

Project 3.  SEQ Project_3. \* ARABIC 17
 Now the most important part. Experiment with various values of step and discover the theorem that the Greeks discovered. What can you say about the ratio of the circumference to the diameter for all circles? 

Project 3.  SEQ Project_3. \* ARABIC 18
You have 100 ft of fencing wire for your garden. If your garden plot is circular, how big a plot can you make? (What is the diameter?)

Project 3.  SEQ Project_3. \* ARABIC 19 A circle is a line with a constant curviture. A spiral is a line with an ever increasing (or decreasing) curviture. Write a variation on the circle program to draw a spiral like the one below. For example:

on spiral

 st

 seth 90

 put 5 into angle

 repeat

  fd 10

  lt angle

  {write your modification here}

 end repeat

end spiral

Areas


In the figure below, the area under the line joining the end points (xOld,yOld) and (xNew, yNew), is the base (xNew- xOld) times the average height (yNew+ yOld)/2. 

We may teach the turtle to compute the area under his path. To do so we will define a new kind of forward command. We will name it fdA which stands for “forward and gather the area underneath”. 

on fdA dist 
--New kind of forward.

 global totalArea
--The global variable totalArea is used to hold the area.

 put xcor() into xOld
--Get the old coordinates.

 put ycor() into yOld

 fd dist 
-- Now do the old forward. 

 add .5*(ycor() +yOld)*(xcor() - xOld) to totalArea   --Add the increment in the area.

end fdA

With this new definition of fdA the turtle will first get his initial coordinates, go forward the given distance (i.e. do the usual forward), and then update the area by adding the area he has just passed over. The variable totalArea is made global so that other procedures can share this variable.  We would like to use this simple tool to calculate the area of arbitrary geometrical shapes. 


We should be a little more careful in calling totalArea the area under the line segment. It is, if we move the turtle from left to right, for then (xcor() - xOld) is a positive number and equal to the base of the quadrilateral. But if the turtle is moved from right to left, (xcor() - xOld) will be negative and so is the negative of the base of the quadrilateral. Therefore the quantity calculated by the turtle in fdA is the negative of the area under the line segment when moving in this direction. This may seem like a nuisance but in fact we will use this signed area to our advantage. 


For example, to calculate the area of the regular pentagon in the figure below we might begin with the turtle at A and move forward to B and then to C. The area under these two line segments are added together by our newly defined fdA above. This area is shaded in the figure on the left. Next we proceed from C to D and back to A. The area under these three line segments is indicated in the figure on the right. Notice also that this contribution to totalArea will be negative since the turtle is moving from right to left. Therefore when we add the two contributions together we are left with the area within the pentagon. 


As a simple example which is easy to check, the following procedure should print the area within a square:

on squareArea

 global totalArea
--This variable must be shared with fdA.

 st

 pu

 setXY 40,150
--Place the turtle anywhere on the screen.

 seth 30
--Give the turtle any heading.

 pd

 put 0 into totalArea
--Initiate the global variable.

 repeat 4 times
--Draw a square 100 units on a side and gather the area.

  fdA 100

  rt 90

 end repeat

 pr totalArea
--Print the total area.

end squareArea

The totalArea should be 10000. Type this program into the editor and change the line setXY 40,150 to place the turtle at some other starting point. Is the area within the square the same? 


This analysis may be applied quite generally to any polygon. Any shape drawn by the turtle of course is a polygon. He always moves in a straight line. If we let the line segments be short enough we may approximate an arbitrary curved shape.


Consider, for example, the figure below. As the turtle moves from A to B to C he gathers positive area. As he moves from C to D to A he gathers the negative area under the figure. When these two contributions are added we are left with the area within the figure.

We may say quite generally that for any simple figure traversed in a clockwise sense that the net area gathered by the turtle in fdA is the area within the curve. If the figure is traversed in the counterclockwise sense, the area gathered will be the negative of the area within the curve. If the turtle were to traverse a symmetric figure eight the net area would be zero. 

Project 3.  SEQ Project_3. \* ARABIC 20 
Determine the area of regular pentagon 100 units on a side. Show that the result is independent of the orientation or position of the pentagon. 

Project 3.  SEQ Project_3. \* ARABIC 21 
Approximate the area of a circle of radius r = 100. To do this you may draw a clockwise polygon with 36 sides (a reasonable number to approximate a circle) whose total perimeter is 2*pi*r. (So if there are 36 steps, what must the length of a single step be?) Fill in the procedure circle below:

on circle r

 global totalArea

 st  
 --Start the turtle.

 ct 
--Clear the text.

 put 0 into totalArea 
--Initiate the variable "totalArea".

 seth 180  
--It will look better this way.

 repeat .......

   {Do the right stuff to draw a 36 sided pentagon which approximates a circle of radius r = 100}

 end repeat 

 pr "Inclosed area = " & totalArea --Print area found.

 pr pi*r^2  
 --Print actual area.

end circle


There are times when the fdA command will not be the best way to get around a figure. Suppose, for example you wanted to compare the two shaded areas in the figure below. 

The best way to make the comparison is to proceed in a figure eight fashion around the four points. The simplest tool to move from point to point is setXY. But if setXY is to gather the area in the same way that fdA did we must modify its definition just as we did before for fd. This new command we will call setXYA and is defined by:

on setXYA pt
                    --A new kind of setXY

 global totalArea
                    --Variable to hold the area.

 put xcor() into xold
                    --Get the old coordinates.

 put ycor() into yold

 setXY pt
                     --Do the old setXY

 add .5*(ycor() +yold)*(xcor() - xold) to totalArea        --Increment the area.

end setXYA 

Project 3.  SEQ Project_3. \* ARABIC 22 Verify that the two shaded areas in the trapezoid above are equal. Can you prove this result using the standard methods of Euclidian geometry? Compare also the unshaded areas. Are they equal as well?

 Chapter  SEQ Chapter \* ARABIC 4

User Defined Functions

Introduction


We have discussed earlier many built-in Transcript functions, for example: xcor(), ycor(), distance(), direction(), and heading(). Commands perform some task—move the turtle, print an answer, calculate a number, etc. Functions on the other hand always return a result. The function xcor() returns the value of the x coordinate of the turtle. The function heading() returns the heading of the turtle. All functions must be followed by parentheses. There may or may not be anything within the parentheses. For example xcor() takes nothing within the parentheses, while distance(30,40) takes two parameters—the x and y coordinates of a point and returns the distance to that point. 


We would now like to define our own functions. As an example consider a function sq which returns the square of a number and the command sq which prints the square of a number:

        
 FUNCTION                              COMMAND


function sq num
on sq num


 return num* num
 pr num*num


end sq
end sq

Notice the difference between this function definition and the command definition. The function definition always begins with the word function while the command always begins with the word on. As a second difference, there will always be a return in a function. The result following return is output by the function to be used by some command. There is also a difference in how these two procedures are used. In the case of the function sq we write:


pr sq(2)

and in the case of the command sq we write:


sq 2

The function requires the parentheses and the command does not. The function sq(2) returns the value 4 to the command pr which acts on the 4 by printing the result to the screen. The command sq 2 simply prints a 4 to the screen. 


You might ask: Why bother with with functions at all? The function and the command seem to do pretty much the same thing. To see the value of the function sq consider the task of finding the hypotenuse of a right triangle given the two legs A and B.


on hypotenuse A, B


 put sq (A) + sq (B) into result
--result is an arbitrary variable.


 pr sqrt(result)


end hypotenuse

The function sq is needed here to return the square of the legs of the triangle to the command put (actually to the add command which then returns the result to put). The command sq wouldn’t do. It would just put the number on the screen.


Remember that when a function is called, the input parameter(s) is enclosed in parentheses: sq(A). (Forgetting these parentheses is a very common mistake. When a program doesn’t run and you see the message “Can’t understand...” check for missing parentheses in your functions.) We see this in the user-defined function sq and in the built-in function sqrt (an abbreviation for square root). (The parentheses are not used in the definition of the function, only when it is called. A little confusing, but you’ll get used to it in time.) 



The reason for the parentheses in the function call has its origin in standard mathematical notation. The generic function is f(x), which reads “f of x”. It says: “Give me a value for x and I will compute a new value and give it back to you”. For example:


f(x) = x2
is the standard form of the square function. In this notation f(2) = 4, f(3) = 9, etc.
 As a peripheral note it should be observed that a number may be raised to a power with the notation x^(power). For example the square of 3 can be written 3^2, which reads: “Three to the power two”. We might also obtain the square root of three with the notation: 3^0.5, which reads: “Three to the one-half power”. 

Project 4.  SEQ Project_4. \* ARABIC 1 Write a function which doubles the input parameter. Fill in the blanks:


function double x


return {etc.}
--Put your code here.


end double

Project 4.  SEQ Project_4. \* ARABIC 2 Write a function which returns the average of two numbers n and m. (There is a built in function average(?,?) but don’t use it here.)


function avg n,m


return {etc.} 
--Put your code here.


end avg

Project 4.  SEQ Project_4. \* ARABIC 3 Here is an algorithm for calculating square roots. Let x be the number whose square root you want. Let x1/2 = n. You want to calculate n. Now x = n2 and so x/n = n. Suppose you made a guess for the value of n. Call the guess “guess”. You have then


where shouldEqualGuess is the quotient of x and guess. If the guess were correct then guess and shouldEqualGuess should be equal; hence the nomenclature. If they are not equal, you might try as a better guess the average of guess and shouldEqualGuess. So here is the algorithm:

Make a guess.

Decide your level of tolerance; how big an error will you allow?

Calculate x/guess = shouldEqualGuess.

If guess is close enough to shouldEqualGuess then stop guessing.

If not, then take as a new guess the average of guess and shouldEqualGuess.

Go back to step 3.

Implement this algorithm to write a function which returns square roots to a tolerance of .001. You will find the Transcript function abs(?) useful. It returns the absolute value of its argument. Why does this algorithm work? 

Project 4.  SEQ Project_4. \* ARABIC 4 Write a function, midPt(P,Q) which returns the midpoint of the line joining the points P and Q. The point must be returned as a list. (You may want to use the function avg defined above or the built in average function.)

function midPt P,Q


 put item 1 of P into px


 put item 2 of P into py


 {etc.}
 
--Put your code here.


 return ...


end midPt

To find the midpoint of the line joining the two points “30,40” and “100,150” you would run 


pr midPt(“30,40”, “100,150”)
 The result should be the list “65,95”. This function, midPt, is very useful. It should be copied (select the text, type command C) and pasted (command V) into the background scripts (option-command B). It will then be available to all card scripts.


We would like to use the function midPt() to solve some problems in geometry. First let us draw a simple triangle. We will use the function clickPoint() which waits for a mouse click and then returns a list of the x and y coordinates of the mouse position.

On triangle

 st

 put clickPoint() into a 
--Wait for a mouse click and then get the mouse coordinates and


 
--put them into the variable a.

 put clickPoint() into b 
--Do the same for b.

 put clickPoint() into c
--Do the same for c.

 lineBetween a,b     
--Draw a line between points a and b.

 lineBetween b,c
--Draw a line between points b and c.

 lineBetween c,a
--Draw a line between points c and a.

end triangle

where we have used the command lineBetween a,b to draw a line between the points a and b. Run triangle, click the mouse at three different points on the screen, and see the resulting triangle. 

Project 4.  SEQ Project_4. \* ARABIC 5 Write a procedure which draws the triangle and then connects the vertices of the triangle with the midpoints of the opposite sides. These lines are called the medians of the triangle.  (Use the procedures: setXY, lineBetween, and midPt.) The result should look something like the figure below.

How does your figure differ from our figure? What general theorem does your figure suggest? 

Project 4.  SEQ Project_4. \* ARABIC 6 Write a procedure which draws the triangle and then a line which connects the midpoints of any two sides. Verify that this line is parallel to the third side. Use direction(?) to determine angles. (See the figure below.)

The line connecting the midpoints is parallel to the base.

Project 4.  SEQ Project_4. \* ARABIC 7 Write a procedure which verifies that the medians of the triangle above meet at a point which divides the medians in the ratio one to two, that is: the long side of a median is twice the short side. You will find direction, midPt, and distance useful. 

Project 4.  SEQ Project_4. \* ARABIC 8 Write a procedure which draws a quadrilateral. Use the function clickPoint() above to generate the four points. Connect the midpoints of the four sides. This will form a quadrilateral within the quadrilateral. What can you say about this new quadrilateral? (See the figure below.)  Use direction(?) to get angles of lines and distance(?) to get lengths of lines. (Choose some convenient notation to identify the midpoints, e.g. put midPt(p1,p2) into m12. You may use label to label vertices.)

A quadrilateral formed by connecting midpoints of a quadrilateral.

Chapter  SEQ Chapter \* ARABIC 5

Turtle Physics

Introduction


We previously considered the problem of a boat crossing a flowing stream. Let us return to that problem. To cross the stream we might use the following cross procedure:

on cross riverVel, boatVel

st

put 0 into riverDirection
--The river flows from left to right across the screen.

put 90 into boatDirection
--The boat is pointed toward the top of the screen.

repeat forever

 seth riverDirection


 fd riverVel
--Let the river flow for one second.

 seth boatDirection

 fd boatVel
--Row the boat for one second.

end repeat

end cross

In this procedure the velocity of the river is riverVel and the velocity of the boat is boatVel. As before, we let the river carry the boat downstream for one second, stop the stream, and then row the boat cross stream for one second. This, of course, is not what happens in nature, but it does give a rough simulation of the real process. 


Suppose the oarsman gets tired as he rows across the river. Suppose the boat velocity decreases by .1 with each second. The crossing procedure then becomes:

on cross riverVel, boatVel

st

setHome -100,0
--Move the origin to the left to give the boatman more room.

put 0 into riverDirection


put 90 into boatDirection

repeat forever

 seth riverDirection

  fd riverVel

  seth boatDirection

  fd boatVel

  subtract .1 from boatVel
--Decrease the boat velocity by the boat acceleration.

end repeat

end cross

Try cross 3,4 and watch the boatman. If you let the procedure run long enough you find that, not only does the boatman not get across the river, he eventually returns to the shore from which he set out. 

Trajectory of a tired boatman.

 This may seem strange, but that is what the procedure gives. If we keep subtracting .1 from the boat velocity, eventually it will become negative and the boat will move backward.


We can smooth the jagged steps with the equivalent procedure:

on cross riverVel, boatVel

st

setHome -100,0

repeat forever

  incXY riverVel, boatVel
--Move the river and the boat together.

 subtract .1 from boatVel

end repeat

end cross

The command incXY increases the x coordinate by an amount equal to the river velocity, and the y coordinate by an amount equal to the boat velocity. This procedure gives us a smooth picture of the crossing.

A smooth crossing.

Projectiles


While this isn’t very realistic for the boat crossing a river (the tired boatman won’t begin to row backward), it is quite realistic for a different physical problem. Suppose a stone is thrown in the air with both a horizontal component of velocity and a vertical component of velocity. Neglecting air friction, the horizontal component of velocity remains constant throughout the flight. The vertical component, however, decreases at a uniform rated due to the pull of gravity. Let us rewrite this procedure to simulate the trajectory of the projectile.

on traj vx, vy

st

repeat forever

 incXY vx, vy

 add accy() to vy

end repeat

end traj

function accy

return -.1

end accy

We have chosen to put the effect of gravity into the function accy. This is the acceleration of the projectile in the y direction. The acceleration is, by definition, the change in the velocity per second. Every second, the y component of the velocity increases by an amount equal to the y component of the acceleration. In this case the y component of the acceleration is negative and equal to -0.1. We will see later the value of removing the acceleration from the command and treating it as a function which returns the acceleration. 

Project 4.  SEQ Project_4. \* ARABIC 9 Confirm the equation for the height of a projectile thrown vertically into the air:

where vy is the vertical velocity of the projectile and g is the acceleration due to gravity (.1 in our example). Instead of using repeat forever, try repeat until vy < 0. Ask to turtle to print his height when he turns at the top. (To simplify the verification you might also ask him for the value of vy2/2g.  Be sure you evaluate vy at the beginning of the procedure, before you begin subtracting the acceleration due to gravity from vy.)


You will find that the confirmation in the above project is not perfect. The reason for the discrepancy is that the procedure is a numerical approximation to the equation of motion. The approximation may be improved considerably with a very simple change. During the first step (incXY vx ,vy) we have ignored the effect of gravity. It isn’t until the step is completed that we added the acceleration. A simple way to improve the approximation in the first step is to add accy/2 to the initial velocity. This way, the velocity during the first step is the average velocity during the first step. If there were an acceleration in the x direction we would also add accx/2. The velocity during the first step is the average velocity, or the velocity at the midpoint of the first step. For this reason this is called the midpoint approximation. On each subsequent step we add the acceleration after one second; i.e. one second after the first midpoint, one second after the second midpoint and so on. Each time we obtain the velocity at the midpoint of the next step. This process is repeated throughout the motion. The projectile is always moving with this average midpoint velocity. Try this modification to traj.

on traj vx, vy

st

add accy()/2 to vy
--The midpoint approximation.

repeat forever

 incXY vx, vy

 add accy() to vy
--The acceleration is always that at the middle of the step.

end repeat

end traj

Project 4.  SEQ Project_4. \* ARABIC 10 Add this midpoint approximation to the previous project for the maximum height of a projectile. Run traj 0,4 and see how much better this midpoint approximation is.

Project 4.  SEQ Project_4. \* ARABIC 11 Modify the trajectory procedure to allow the projectile (ball) to bounce when it hits the ground, that is when ycor() is less than zero. (If your ball does not rebound to the same height with each bounce check to be sure that the next step after reaching the ground—actually a point just below the ground—is exactly the reverse of the preceding step. Do not add the acceleration. You may find next repeat or if—then—else useful. You may also wish to use setHome to move the origin to the left so that there is more room to bounce.)

It would be more useful if the projectile procedure took the speed and direction as inputs rather than the x and y components of the speed. For example, we would like to see what happens to a projectile thrown with velocity of 6 ft/sec at an angle of 60 degrees. We may do this as follows:

on projectile velocity, direction

  st

  seth direction
--Point the turtle in the direction of the throw.

  fd velocity
--Move the turtle out a distance equal to the velocity.

  put xcor() into vx  
 -- The x coordinate of the turtle will be vx.

  put ycor() into vy   
-- The y coordinate of the turtle will be vy.

  traj vx,vy
--Now that we have the velocity components we may use traj.

end projectile

We have used the turtle to find the x and y components of the velocity. The turtle sets his heading to the direction of the throw. He goes forward a distance equal to the velocity. At this point, the x and y components of the velocity are just the x and y components of the turtle as illustrated in the figure below. 

The velocity components are obtained from the turtle’s xcor() and ycor().

Project 4.  SEQ Project_4. \* ARABIC 12 Modify the above projectile procedure to show the ground level and to print out the range (the net horizontal displacement) when the projectile hits the ground, i.e. when ycor(0) is less than zero.
Project 4.  SEQ Project_4. \* ARABIC 13 At what angle should a ball be thrown to achieve maximum range for a given velocity? The velocity is fixed but the angle is variable. Try several angles, and find that one which gives the greatest range. 
Project 4.  SEQ Project_4. \* ARABIC 14 Confirm the following equation for the maximum range: 

where R is the range, g the acceleration of gravity (.1 in our example) and v is the velocity. Since this is the maximum range, the angle should be that which you found in the project above (45o). (Use the midpoint approximation. Even with this correction, the agreement will not be that good. The reason is that the turtle actually falls below ground level before he can be stopped. The are ways to fix this but we won’t take the time. You might try if you like.)

Projectile Motion II


If the change in velocity (the acceleration) is large in comparison to the velocity itself, the numerical algorithm we have been using breaks down. We have always assumed that the velocity is approximately constant during the step—fd vx or fd vy. If the acceleration is large, the change in velocity is large.  The midpoint approximation will help, but in many examples the velocity will vary in a nonuniform way during the step. The simplest way to deal with this case is to shorten the time interval over which a step is taken so that the change in velocity over this smaller time interval is small. 


We have always taken the time interval to be one second. During one second, the object will move a distance equal to the velocity (the number of feet moved per second). Also, during one second, the velocity will change by an amount equal to the acceleration (the change in velocity per second). If the time interval between steps is say dt seconds then the distance moved along the x axis will be vx * dt—the distance moved per unit second (vx) multiplied by the number of seconds (dt). (Think of dt as the difference in time between two successive step.)


In exactly the same way, the change in the x component of the velocity during a time dt will be accy * dt—the change in velocity per second (accy) multiplied by the number of seconds (dt).


Let us return to our projectile procedure and shorten the time interval between steps from one second to a time difference dt. We will call the procedure throw:

on throw velocity , direction

  st

  setHome -200,-100 
-- Move the origin 200 pixels to the left and 100 pixels down.

  seth direction
--Point the turtle in the direction of the throw.

  fd velocity
--Move the turtle out a distance equal to the velocity.

  put xcor() into vx   
--Put the components of the velocity into vx and vy

  put ycor() into vy   

  bk velocity      
--We have the components now so go back home.  

  put .05 into dt    
--The time interval between steps is taken to be .05 seconds.

  add accy()/2*dt to vy  
-- We are using the midpoint approximation again.

  repeat forever

   incXY vx*dt, vy*dt  
-- The increment in the x coordinate during the time dt is vx*dt.

   add accy()*dt to vy   
-- The increment in vy during the time dt is accy*dt.

  end repeat

  pr "Range = " & xcor()
-- The symbol “&” is used to join two strings of text.

end throw

function accy

  return -32 
-- This is the acceleration of gravity on Earth.

end accy

 The throw procedure now allows us to solve more realistic problems. We can make the time dt small enough to handle large velocities. For example if dt = .05 and v = 80 ft/sec, the distance moved on the screen will be .05 * 80 = 4 pixels. This is a good compromise between accuracy of the calculations and the speed at which the program runs.  The acceleration of 32 ft/sec/sec is the acceleration due to gravity on Earth. The symbol “&” is called the ampersand, and is used to concatenate (join) two strings (phrases). (You may use two ampersands--&&-- to join two strings and leave a space between them.) The pr statement will print “Range = ” and then the x coordinate (the range) of the turtle.

Project 4. SEQ Project_4. \* ARABIC 15 A baseball pitcher can throw a baseball at 88 ft/sec (60 mph). What is the maximum distance he could throw the ball neglecting air friction? (The turtle may go off the screen. Be patient. He will land eventually.)

Project 4. SEQ Project_4. \* ARABIC 16 If you could throw a baseball with a speed of 88 ft/sec, how far could you throw it on Jupiter where the acceleration of gravity is 2.56 times that on Earth?

Chapter  SEQ Chapter \* ARABIC 6

The Harmonic Oscillator

Introduction


In many physical systems the acceleration is not a constant. For example an harmonic oscillator (a weight oscillating at the end of a spring), a pendulum (a weight tied swing from the end of a string), planetary motion (the motion of the planets about the sun) are examples in which the acceleration is not a constant. In this chapter we will examine the harmonic oscillator. Two simple examples of the harmonica oscillator are illustrated below. 

The first is a block oscillating at the end of a spring. The second is a simple pendulum. In each case we find that the acceleration of the mass depends on its position. In fact, we find for small displacements from the equilibrium position that the acceleration is proportional to the displacement from the equilibrium position. That is:

a = - w2 x

where w is a constant and x is the displacement from the equilibrium position. (We will see presently why we chose the constant to be of the form w2 .) 


From this equation we see that when the displacement is positive the acceleration is negative and when the displacement is negative the acceleration is positive. This means that as long as x is positive the velocity will decrease. Therefore if x if positive and the velocity is decreasing, then the velocity must eventually become negative, and so turn around and head back to the equilibrium position where x is zero. Similarly, if x is negative and the velocity is increasing, then the velocity must eventually become positive, and so head back to the origin. But this is just what we would expect in something that oscillates. In general we would expect that any functional relationship between acceleration which satisfies the condition of reversing sign at some point would exhibit oscillatory behavior. What is special about harmonic oscillation is that the relationship between acceleration and displacement is linear. This is illustrated in the figure below.

Notice that for any oscillator that for small displacements, small values of x, we may approximate the curve by a straight line—the dotted line in the figure above. Thus the study we will make for harmonic oscillators will be a good approximation to all oscillators if the amplitude is small.

The One Dimensional Harmonic Oscillator


It is a simple matter to write a program to simulate an harmonica oscillator. 

The turtle moves up and down along a verticle line. 


It’s hard to see exactly what is happening and so let us make a plot of the of the oscillator as a funciton of time. We may do this as follows:

on oscillator amplitude

 global w

 st

 pu

 put 1 into dt

 put .1 into w

 setY amplitude --Set the oscillator at the top position.

 pd

 put 0 into vy --The osciallator is release from rest, i.e. vy = 0

 repeat until vy > 0

  incXY dt,vy*dt

  add accy()*dt to vy

 end repeat

end oscillator

function accy

 global w

 return -(w^2)*ycor()

end accy
This will show us a plot someting like that below

Project 6.  SEQ Project_6. \* ARABIC 1 Verify that the period of oscillation, the time betweeen osciallations, is given by T = 2 pi/w. Hints: Introcude a time variable, add dt to the time with each step, and use repeat until vy > 0. After the repeats let the program print out 2*pi/w and 2 * time. You can inprove the approximation by incorporating the midpoint correction (add accy()*dt/2 to vy).

Project 6.  SEQ Project_6. \* ARABIC 2 How is the period of scillation related to the amplitude? Try different amplitudes for a fixed w. Why does this make the harmonic osciallator a good clock? Remember that a clock’s amplitude will depend on how tightly it is wound.

The Two Dimensional Harmonic Oscillator


The twodimensional harmonicc oscialltor is very rich in the kinds of motion it displays. Two examples of harmonic oscillators are illustrated below:

Examples of two dimensional oscillators

In the first example a weight is tied to four springs and slides without surface friction in the horizontal plane. In the second example a weight is tied to a string and rotates in an ellipse without air friction. To obtain the motion of these two harmonic oscillator we employ the following:

on osc vx,vy

 st

 label “x” 
--Locate the center of oscillation.

 pu

 setXY 100,0
--Pull the weight out 100 pixels to the right.

 pd

 add accx()/2 to vx    
--The midpoint approximation.

 add accy()/2 to vy

 repeat forever

  incXY vx,vy

  add accx() to vx

  add accy() to vy

 end repeat

end osc

function accx

 return -.01*xcor()    --The x acceleration is proportional to the x displacement.

end accx

function accy

 return -.01*ycor()    --The y acceleration is proportional to the y displacement.

end accy

Notice that the acceleration depends on the distance of the weight from the point of equilibrium (where the “x” is labeled). For the harmonic oscillator this acceleration is directly proportional to this distance from the equilibrium position. The factor of proportionality depends on the stiffness of the springs. In the above example this factor has been chosen to be .01 for all springs. When the four springs are of the same strength the motion will be an ellipse. Try osc 0,4 to see an ellipse.

Project 6.  SEQ Project_6. \* ARABIC 3 This is a more difficult project. Prove that the path of the two dimensional harmonic oscillator above is an ellipse. To do this you must prove that the sum of the distances between the foci of the ellipse and any point P on the ellipse is a constant. You will have to find the length of the major axis (2a) and the minor axe (2b) and, from these, the position of the foci. Ask the turtle to calculate u and v and print the sum as he moves around the ellipse. The result should be a constant. 

a is the semimajor axis

b is the semiminor axis

u + v = 2a for any point P on the perimeter of the ellipse


If the strength of the springs is not the same, the restoring force in the x direction with not be the same as that in the y direction. We might have something like the following:

function accx

 return -.02*xcor()

end accx

function accy

 return -.01*ycor()

end accy

In this case the motion will be more complicated. It might look something like the following:

A Lissajou Figure

Such figures are called Lissajou figures. (They will form a closed figure if the ratio of the square of the accelerations is a rational number.) 

Project 6. SEQ Project_6. \* ARABIC 4 Most watches use an harmonic oscillator of some sort--a spring, a pendulum, a tuning fork, etc. To see why a harmonic oscillator is a good time keeper, use the following procedure for a one dimensional harmonic oscillator to see how the period depends on the amplitude:
Predator-Prey Relations


Periodic motion is not limited to physical systems. Many natural systems exhibit this behavior. One of the most interesting examples is what biologists call “predator-prey systems.” It deals with the ecological balance of natural systems. As a special example of this theory let us imagine an island on which there are only two animals, rabbits and foxes. The rabbits eat grass and the boxes eat rabbits. In order to study the population dynamics of this system we ask the following question: If there are currently 80 rabbits and 40 foxes on the island, how many rabbits and foxes will there be next year, the year after than, and so on? 


Of course we need some more informatin to solve this problem. Let us use R to represent the number of rabbits and F the number of foxes. We asume that the growth rates for therabbits and foxes satisfy the following equations (called the Volterra equations):

Rate of increase of R = a R - C R F

Rate of increase of F = -b F + d R F

The first term on the right hand side of each equation represents the natural growth rate of each species in the absence of the other. If there were no foxes, the rabbits would grow at the rate of a R (assumming unlimited grass). The coefficient a is the growth rate per rabbit per year of the rabbit population. If the growth rate per rabbit is multiplied byteh number of rabbits (R) we get the growth rate of the entire population. In a similar way, ifthere were no rabbits to feed on, the foxes die out at a rate -b F.

The second term in each equation represents the number of encounters between rabbits and foxes. These encouters (not pleasant to behold) will contributretothereduction ofthe rabbit population ad the nourishment of the fox poppulaiton. Now the number of encounters between rabbits and foxes will be proportional to the product R F. For example, if you double thenumber of rabbits or the number of foxes you will double the number of encounters.


The Volterra equations can be solved using the preditorPrey program below.




on preditorPrey r,f

 setHome -200,-100

 drawAxes

 put .04 into a

 put .03 into b

 put .0004 into c

 put .0004 into d

 pu

 setXY r,f

 pd

 repeat until the mouse is down

  add a*r - c * r * f to r

  add -b*f + d * r * f to f

  setXY r,f

 end repeat

 choose the browse tool

end preditorPrey

on drawAxes

 fd 300

 label "Rabbits"

 pu

 home

 pd

 seth 90

 fd 300

 label "Foxes"

end drawAxes

To see this in actin try preditorPrey 30,40 that is, the evolution of 30 rabbits and 40 foxes. You should see something like the following figure:

Project 6. SEQ Project_6. \* ARABIC 5  By experimenting with different inital values of R and F see if youcan determine the “equilibrium state” of the system; that is, those initail valuses of R and F for which there is no change with time. Can you see why these are the equilibrium values by examining the rate equations for rabbit/fox populations?

Chapter  SEQ Chapter \* ARABIC 7

The Monkey and the Hunter

Einstein’s Principle of Equivalence

Introduction

In this chapter we will apply Einstein’s principle of equivalence to solve a classic classroom demonstration in physics: The monkey and the hunter. In this experiment a gun is aimed at a monkey sitting on a tree limb. The gun is fired and, at the same instant, the monkey drops from the limb. (Of course in class we use a metal tube with a pellet for the bullet and a tin can suspended from an electromagnet for the monkey.) The question is: What happens? Does the bullet pass above the monkey, below the monkey, or does it strike the monkey? To answer this question we need to be able to write programs for the motion of the bullet as well as the monkey. We need two turtles. 


To deal with this problem and others like it we will introduce a second animated object, the monkey. To talk to the monkey we must first start the monkey in much the same way we started the turtle. To initiate the turtle we use st (for start turtle). In the same way we initiate the monkey with sm (for start monkey). We talk to the turtle just as we have been. To talk to the monkey however we must bracket his instructions with tellMonkey and endMonkey. This is very similar to the kind of bracketing we use with repeat and end repeat, or if and end if. 


Let us consider first a simple example of multiple animals. Imagine a monkey chasing a turtle who is running in a circle. The monkey always runs directly toward the turtle. This is accomplished in chase below: 

on chase

 st
--Initiate the turtle.


 sm
--Initiate the monkey.

 put 6 into velTurtle
--The turtle’s velocity is 6.

 put 5 into velMonkey
--The monkey’s velocity is 5. (We didn’t promise realism.) 

 pu

 setxy 100 ,0
--The turtle starts at 100,0. The monkey will start at 0,0.

 pd

 repeat forever

  fd velTurtle

  lt 5   
--The turtle turns through a constant angle and so moves in a circle.

  put xyCor() into TurtlePt
--Locate the turtle.

  tellMonkey
--Start talking to the monkey.

    seth direction(TurtlePt)
--The monkey faces the turtle.

    fd velMonkey
--Move the monkey toward the turtle.

  endMonkey
--Stop talking to the monkey.

 end repeat

end chase

Project 7.  SEQ Project_7. \* ARABIC 1 Calculate (with pencil and paper) the radius of the turtle’s circle and that of the monkey in the above example. Verify that these are the radii of the turtle’s and monkey’s circles in the above chase.

The Monkey and the Hunter

An insensitive hunter is out to bag himself a monkey. He spots a monkey siting on the branch of a tree. Taking careful aim he fires the gun directly at the monkey. The monkey sees the flash of the gun and instantly drops from the branch. What happens? Does the bullet pass above, or below the monkey. To answer the question run monkeyHunter. You may set the velocity of the bullet and the angle at which the gun is aimed using the velocity and angle parameters. Since the monkey is located at the xyCor (200,200), the line of sight from the hunter to the monkey is 45 degrees. 

on monkeyHunter velocity,angle

 st
--Initiate the turtle.

 sm
--Initiate the monkey

 setHome -150,-30
--Move the origin 150 to the left and 30 units down.

 fd 250
--Draw the ground.

 bk 250

 put "200,200" into monkeyPoint
--Place the monkey at 200,200

 put velocity into vBullet

 seth angle
--Direct the turtle in the direction of the gun.

 fd vBullet
--Get the components of the bullet’s velocity.

 put xcor() into vBulletx
--The x component of the bullet’s velocity.

 put ycor() into vBullety
--The y component of the bullet’s velocity.

 bk vBullet
--Go back to the origin.

 put 0 into vMonkey
--The monkey is sitting at rest on the branch.

 put -.1 into accGravity
--The acceleration of gravity.
 

 tellMonkey
--Start talking to the monkey.

 pu
--Pick up the pen.

 setXY monkeyPoint
--Place the monkey on the branch.

 pd
--Put the pen down.

 label “x”
--Put an x at the point from which the monkey falls.

 endMonkey
--Stop talking to the monkey.

 repeat until xcor() >200
--Continue until the bullet passes the monkey’s fall line.

  incxy vBulletx,vBullety
--Move the bullet.


  add accGravity to vBullety
--Add the bullet’s acceleration due to gravity.

  tellMonkey
--Start talking to the monkey.

    incXY 0,vMonkey
--Move the monkey.

    add accGravity to vMonkey
--Add the monkey’s acceleration due to gravity.

  endMonkey
--Stop talking to the monkey.

 end repeat
--Continue the cycle.

end monkeyHunter

Fire the gun above the monkey (say 46 degrees), below the monkey (say 44 degrees) and directly at the monkey (45 degrees). You should find that the bullet fired directly at the monkey will strike him. The bullet should strike the monkey regardless of the muzzle velocity. 

The hunter aims the gun directly at the monkey.

If the velocity of the bullet is high, it will strike the monkey close to the branch. If the velocity of the bullet is low, it will strike the monkey close to the ground. Try a few different muzzle velocities.

How do we account for this coincidence? No matter what the muzzle velocity of the bullet, it always strikes the monkey. To make matters even more puzzling, change the acceleration due to gravity from -.1 to -.05. Fire the bullet at the monkey again. Again the monkey is struck regardless of the velocity of the bullet. If we were to set the acceleration of gravity equal to zero we can easily understand the results. With no gravity the monkey doesn’t fall and the bullet moves in a straight line to strike the monkey. No puzzle there.


To understand these results let’s look at the experiment from the point of view of a person who fells no gravity. Imagine a person standing in an outdoor elevator. (Use your imagination.) The minute the gun is fired the cable supporting the elevator is cut and it begins to fall freely. The moment the cable is cut the person in the elevator feels weightless. He loses all sensation of gravity. If the elevator were totally enclosed, he might just think he has suddenly been transported to outer space where there is no gravity. In fact we are all aware that the astronauts have no sensation of gravity. They feel weightless. If they were to throw a ball, it would move in a straight line relative to them. Of course, to someone on the earth the ball would move in a arc, much the same way the space ship does. 


These phenomena are examples of Albert Einstein’s principle of equivalence, which states:

It is impossible to distinguish between an accelerated frame of reference and a uniform gravitational field.

The man in the elevator and the astronauts in the space ship are in accelerating frames of reference--a very special frame: a reference frame which is falling freely under gravity. The acceleration is just right to cancel the effect of gravity. 


The laws of physics must apply in all frames of reference. In a frame of reference in which there is no gravity, monkeys do not fall out of trees, and bullets travel in a straight line. To the man in the elevator, the ground and the tree are accelerating upward toward him but the monkey remains at rest and is struck by a bullet traveling in a straight line. It is a very simple problem as far as he is concerned. Both the man in the elevator and the hunter must see the same result. If one sees that the monkey is struck, the other must see the monkey struck as well. Each sees the gun pointed directly at the monkey and each sees that the monkey is struck by the bullet regardless of the bullet’s velocity. The only difference is their interpretation. 

The trajectory of the bullet and the falling

 monkey are viewed by two different observers.

To the hunter, the bullet travels in a parabolic path and strikes the monkey some distance below the limb. To the man in the falling elevator, the bullet travels in a straight line and strikes a monkey who is at rest.

Chapter  SEQ Chapter \* ARABIC 8

Planetary Motion

Newton’s law of universal gravitation

We have studied the motion of projectiles. In a sense, a projectile is very similar to planetary motion. The planets move about the sun because the sun exerts a gravitational pull on the planets and keeps them from wandering off into space. In the same way, the earth exerts a gravitational pull on a stone and keeps it from being flung off into space.


The problem of gravitational forces had puzzled scientists for over two thousand years. To the Greek astronomers there were many question to be answered. Why does a stone fall to the ground? (Aristotle’s answer to this question was: That’s where it belongs. That is its natural state. That is better philosophy than physics.) Why does the Moon rotate about the earth? Is there any connection between the motion of the Moon about the Earth and the trajectory of a stone? If so, why doesn’t the Moon just fall to the Earth the way a stone does?


Newton recognized the connection between the motion of a projectile and the motion of planetary bodies. In his treatise Principia-Mathematica, we find an illustration very similar to that of figure below.

A ball is thrown from a tall pole with ever increasing velocity.

The figure represents a body being thrown from a very tall pole (the North Pole?). For low velocities the trajectory is just what we would expect. As the velocity increases, however, the point at which the body strikes the Earth is further and further from the pole. It seems quite reasonable that at some very high velocity it will return to the pole without ever striking the ground. Perhaps this is precisely how the Moon behaves. It has a velocity which permits it to “fall” in the Earth’s gravitational field in such a way that its orbit is a circle. To verify this conjecture Newton need a quantitative law that he could test mathematically. The law he proposed was the law of universal gravitation which states that:

Any two bodies are attracted to each other with a force which is proportional to the product of their masses and inversely proportional to the square of the distance separating the centers of the two bodies. The force is attractive and is directed along the line joining the two bodies. Or in mathematical form:

where F is the force on either body, m and M are their respective masses and r is the separation between the mass centers. This law applies to all bodies. Your body is attracted to the Earth with a force proportional to you mass and the mass of the Earth. If you body were on the Moon, your mass would be the same but the mass of the Moon is much less than that of the Earth and so the gravitational attraction between your body and the Moon would be much less. Therefore, you would weigh less on the Moon.


You can measure this force of attraction between yourself and the Earth by putting a spring under your feet and measure its compression. This is essentially what you do when you weigh yourself on a bathroom scale.


In order for the gravitational force to be large there must be at least one large mass. Although there is a gravitational force between you and your house, it is very small. However, with very sensitive instruments, it is possible to measure the attractive force between any two bodies of known mass and so measure the factor of proportionality in the above equation. The constant is called the universal gravitational constant and the symbol G is used. The gravitational force becomes then:

where G has the measured value of 6.67 x 10-11 in metric units. 


This law must be combined with Newton’s law of motion. This law is:



F = ma

Putting these two equations together: 

or since the mass m is common to both sides of the equation:

(The Inverse Square Law)

This is the law we need to determine the motion any planetary body. It tell us the acceleration given the mass of the attracting body and the distance to that body. (Notice that the acceleration of a given body depends only on the mass of the other body.) It is often called the inverse square law.


On of the simplest applications of this formula is to a body dropped near the surface of the Earth. Substituting G = 6.67 x 10-11, M = 5.97 x 1024 kg, and r = 6.37 x 106 m, we find a = 9.8 m/sec/sec or equivalently a = 32 ft/sec/sec, which is the observed acceleration due to gravity at the surface of the Earth. On the Moon, a = 5.1 ft/sec/sec and on Jupiter a = 85 ft/sec/sec. If you fell out of bed on Jupiter you might break your pajamas. 

The turtle goes into orbit

To determine the path of a planetary body we can return to our trajectory procedures and simply add the new form for the acceleration. Then, if we had a screen as big as the solar system, we could draw the planets’ paths. Lacking the large screen we must make some adjustments to keep the turtle in view. (We will deal with realistic numbers for the solar system a little later.) These compromises have been made in the procedure orbit:

on orbit vx,vy

 st
--Initiate the turtle.

 label “x”
--Draw an x at the position of the Sun.

 pu

 setx 150
--Put the orbiting body at 150 pixels.

 pd

 add accx()/2 to vx
--The midpoint approximation.

 add accy()/2 to vy

 repeat forever
--Start the orbit.

  incXY vx,vy
--Move the body.

  add accx() to vx
--Add the acceleration to the velocity.

  add accy() to vy

 end repeat

 end orbit

function accx

 --The x component of acceleration from Newton’s law.

 return -4000*xcor()/distance(0,0)^3  --Set GM = 4000

end accx

function accy

 --The y component of acceleration from Newton’s law.

 return -4000*ycor() /distance(0,0)^3  --Set GM = 4000

end accy

Most of this program is the same as that for a projectile. The main difference is the acceleration function. A word must be said about this form. It doesn’t look much like the inverse square law for the acceleration law that we would expect. 


The factor of 4000 is simply a convenient scale factor. It keeps the turtle on the screen. The remaining factors are the components of the inverse square acceleration. To see this let us look at the y component of the acceleration. Consider the figure below showing the the Sun and the Earth. (For simplicity we show the acceleration directed away from the Sun. It should be toward the Sun. We will fix this in the end by reversing the sign.) 

Notice the two similar triangles. From these we have the relation:

where ycor() is the y coordinate of the Earth, distance(0,0) is the distance of the Earth from the Sun, accy() is the y component of the acceleration of the Earth, and finally, acc is the acceleration of the Earth. Solving for accy() we have:

Now the acceleration of the Earth is given by the inverse square formula. So that:

where r = distance(0,0) and GM = 4000 in our example. Putting these equations together:

Now, as we mentioned above, we have the sign reversed. The Sun does not repel the Earth. With this change of sign and setting GM = 4000 as a convenient scale we have:

which is the acceleration function employed in our program. A similar analysis applies to accx(). (We could have used the turtle to calculate the components of the accelerations as we did in the throw procedure earlier.)


Try orbit 0,4. You should see an elliptical orbit. Perhaps it is better to say an oval orbit. An ellipse is a very special mathematical figure. Before we can say that the orbit is an ellipse we must do some more work. In fact what we would like to do is to prove Johannes Kepler’s three laws of planetary motion.

Project 8.  SEQ Project_8. \* ARABIC 1 Most the the planets in the solar system move in orbits which are almost circular. Determine the relationship between the velocity and radius in order for the orbit to be circular. To find this relationship it is helpful to modify orbit a bit. We will call the new procedure circularObit:
on circularOrbit vy , r

 st
--Initiate the turtle.

 label “x”
--Draw an x at the position of the Sun.

 pu

 setX r
--Put the orbiting body at this initial distance from the sun.

 pd

 put 0 into vx

 add accx()/2 to vx
--The midpoint approximation.

 add accy()/2 to vy

 repeat until xcor() < 0  
--Stop when the planet reaches the top of its orbit.

  incXY vx,vy
--Move the body.

  add accx() to vx
--Add the acceleration

  add accy() to vy

 end repeat

 pr r && distance(0,0) 
--Print the original and the final distance from the Sun. These 
--should be equal in a circular orbit.

end circularOrbit

function accx

{the same as before}

end accx

Run circularOrbit 4, ? to find the distance at which the orbit will be circular if the velocity is 4. Next double the speed to 8. At what new radius is the orbit circular? Use this result to select the correct relationship between the r and v for circular orbits. Choose among the following options:


(a) rv2 = a constant

  
(b) r2v = a constant


(c) rv = a constant


(d) r/v = a constant

Show also that the constant is GM (4000 in our example). 

Project 8.  SEQ Project_8. \* ARABIC 2 Write a procedure which automatically produces a circular orbit for any initial distance from the sun. Use the results of the project above to fill in the blanks:

on autoCircular r

 st
--Initiate the turtle.

 label “x”
--Draw an x at the position of the Sun.

 pu

 setx r
--Put the orbiting body at this initial distance from the sun.

 pd

 put 0 into vx

put (?) into vy
--Choose that velocity from (a,b,c,d) above which gives a 
--circular orbit.

...

{The rest is the same as before.}

end autoCircular

You should be able to run autoCircular for any radius and obtain a circular orbit.

Kepler’s three laws of planetary motion

Tycho Brahe, a sixteenth century, Italian astronomer spent his life making exhaustive, detailed measurement of the planetary orbits. It was left to Johannes Kepler to organize the massive detail in Brahe’s tables into three empirical laws of planetary motion. These laws are:

The planets move in elliptical orbits with the sun at one focus.

 A line joining the sun and a planet sweeps out equal areas in equal times.

 The square of the planet’s period is proportional to the cube of the major axis of its orbit.

We shall leave it as an exercise to verify these three laws.

Project 8.  SEQ Project_8. \* ARABIC 3 Prove that the planetary orbits are elliptical. You may find the following steps useful:

 Modify orbit so that you may determine the position of the other focus. (One focus, 
of course, is the sun.)

 Modify orbit again so that it prints out the sum of the distances between the planet 
and the two foci after each step.

 Run orbit 0,4 with these modifications. The sum should vary by only a few percent. (If necessary, change the print field to a scrolling field to see the results.)

Project 8.  SEQ Project_8. \* ARABIC 4 Show that a line from the sun to the planet sweeps out equal areas in equal times. In the figure below, orbit has been modified to show lines drawn from the planet to the Sun after equal time intervals. It is at least plausible that the area of these triangle-like regions are constant throughout the orbit. 

To prove this you must be able to calculate these areas. Consider first the area of a circular segment as illustrated in the figure below:
The area of the shaded figure on the left if just the fraction J/360 of the area of the entire circle. So that

However, the area of the segments shown in the planetary orbit are not circular. The radii of the segment are not the same. Such a case is illustrated in the right side of the figure above. The area on the shaded figure on the right can be written approximately:

where we have used the averaged radius of the circular segment. To obtain these two radii (as well as the angle) we might use the following code:

on orbit vx,vy

{etc}

repeat forever

 seth direction(0,0)

 put heading() into oldHeading

 put distance(0,0) into oldRadius

 incXY vx,vy

 seth direction(0,0)

 put heading() into newHeading

 put distance(0,0) into newRadius

 {Include steps necessary to print the area swept out during this cycle.}

end repeat

end orbit

There is a somewhat more efficient way of doing this, but it will do the job. Run orbit 0,4 with these modifications and observe the printout of the areas. The approximated areas shouldn’t vary by more than a few percent.

Project 8.  SEQ Project_8. \* ARABIC 5 Show that the ratio T2/(major axis)3 is approximately constant. In this ratio, T is the period of the planetary orbit. Modify orbit so that only one half of the orbit is drawn. This will allow you to determine the major axis and half the period. The period is just the total number of steps taken to complete one orbit. Run orbit vx,vy with vx = 0 and vy = 4,4.5,5,5.5 and compare the ratio T2/(major axis)3 for these four cases. 
Chapter  SEQ Chapter \* ARABIC 9

Music of the Spheres

SCALING

Up to this point we have considered only imaginary orbits. We would now like to consider the problem of simulating the motion of Mercury, Venus, Earth, Mars, Jupiter, Saturn, Uranus, and Pluto. The biggest problem in doing this is one of scale. Its not just scaling the orbits to the screen size. We can easily scale to orbit of Pluto so that it fits on the screen. However the orbit of Mercury is one hundred times smaller than that of Pluto. If we scale the planetary orbits so that the radius of Pluto’s orbit is say 100 pixels, then Mercury’s radius would be 1 pixel--clearly too small to be seen or drawn accurately on the screen. We will therefore do the planets in two groups; Mercury, Venus, Earth, and Mars followed by Jupiter, Saturn, Uranus, Neptune, and Pluto.


We can no longer use xcor() and ycor() to keep track of the planet. We will denote the coordinates of the planet by x and y. If this planet is Mars, x and y will be of the order of 2.3*1011 meters, the radius of Mars’ orbit. Xcor() and ycor() of course are screen coordinates and are of the order of 150 pixels. The coordinates x and y must be scaled to fit on the screen. Let us choose a scale so that Mars will fit nicely on the screen at 150 pixels. We define the scale to be the number of pixels per meter, so that 

We will be making our calculations without regard to the scale at all. There is no need to be concerned with the scale when calculating the acceleration or adding the velocity increment to the x and y coordinates. The only time the scale comes into play is when something is done on the screen. It is the screen that must be scaled not the planets. Therefore, when using setX, or setXY the input parameters must be scaled. So that we say for example: setX r*scale, where r is the radius of the planetary orbit in meters and scale is the number of pixels per meter. The product of meters and pixels per meter gives us a screen distance in pixels.


We will also have to make an adjustment in our time interval dt or the orbit will take too long to develop. We will do this by adjusting the length of the turtle step to be some reasonable screen distance. The distance the turtle moves on the screen is v*dt*scale where v is the planetary velocity. On the first step the planet will have only a y component of velocity so that on the first step the turtle moves a distance vy*dt*scale. If this distance is too small, the orbit will take too long to develop. If the distance is too large, the orbit will not be very accurate since our approximation assumes that the velocity is fairly constant over the length of the step. As a reasonable compromise we set:


vy * dt * scale = 4

or 


dt = 4/(vy * scale)

This should be a reasonable time interval between step.


The other new feature of orbit is that we wish to examine only the planetary orbits. These orbits are very nearly circular. We have seen above in one of the projects that the relationship between the velocity of a planet and its orbit radius is:

if the orbit is circular. We may therefore force the orbit to be circular by setting the velocity to this value.

Real Orbits

 We may put all this together in the orbit procedure below.

on orbit r

 global GM
--This variable will be used in the acceleration functions.

 put 2.0*10^30 into massSun

 put 6.67*10^-11 into G

 put G*massSun into GM

 put 150/Mars() into scale
--With this scale, Mars will have an orbit of 150 pixels.

 put r into x
--The planet is located at a distance r from the sun.

 put 0 into y
--And it is on the x axis.

 put 0 into vx
--Its initial velocity is perpendicular to the x axis. 

 put sqrt(GM/r) into vy
--This will ensure a circular orbit.

 put 4/(vy*scale) into dt        --The length of the first step is set to 4. Not too big, not too small.

 pu

 setX r*scale
--Put the turtle at r*scale.

 add accx(x,y)*dt/2 to vx
--The midpoint approximation.

 add accy(x,y)*dt/2 to vy

 pd

 repeat until y < 0
--Do only one half the orbit.

  setXY x*scale, y*scale       --Place the turtle at the current scaled coordinates.

  add vx*dt to x
--Increase the x coordinate by an amount vx*dt

  add vy*dt to y
--Same for the y coordinate.

  add accx(x,y)*dt to vx
--Increase the x component of the velocity by accx*dt.

  add accy(x,y)*dt to vy
--Same for the y component of the velocity.

  end repeat
--Repeat the loop.

end orbit

function accx x,y

 global GM

 return -GM*x /(x^2 + y^2)^1.5
 --Newton’s law of universal gravitation. 

end accx

function accy x,y

 global GM

 return -GM*y /(x^2 + y^2)^1.5
  --Newton’s law of universal gravitation.

end accy

To see the orbit of Earth run orbit 2.3*10^11, where 2.3*10^11 is the radius of the Earth’s orbit in meters. You may compare the Earth’s orbit the that of the other planets by simply choosing a different value of r as the input to orbit. To save time we have included below nine functions which return the radius of the orbit of the given planet. To see the Earth’s orbit then you may run orbit Earth(). To see Mars run orbit Mars(). 

Average radii of the planetary orbits



function Mercury

 return .58 * 10^11

end Mercury


function Mars

 return 2.3 * 10^11

end Mars
function Uranus

 return 2.97 * 10^12

end Uranus

function Venus

 return 1.08 * 10^11

end Venus
function Jupiter

 return 7.8 * 10^11

end Jupiter
function Neptune

 return 4.5 * 10^12

end Neptune




function Earth

 return 2.3 * 10^11

end Earth


function Saturn

 return 1.43 * 10^12

end Saturn
function Pluto

 return 5.9 * 10^12

end Pluto


We have mentioned above that it is not possible to put all planets on the screen at once. To see the outer planets, those beyond Mars, it is necessary to change the scale. Change the line: 


put 150/Mars() into scale 

to the line:

 
put 150/Pluto() into scale

and try orbit Saturn() or orbit Uranus() etc. 

Project 9.  SEQ Project_9. \* ARABIC 1 Determine the period of rotation for all nine planets.  Modify orbit so that it will print the period, measured in Earth years, after one half of the orbit. (The time dt is measured in seconds.) (The observed periods are in order: .24, .61, 1.0, 1.9, 12, 29, 84, 165, 248. Your results will differ somewhat from these values in part because the calculation is only approximate and in part because the planetary orbits are not exactly circular.)

Project 9.  SEQ Project_9. \* ARABIC 2 Modify the procedure orbit to determine the period of rotation of the Moon about the Earth. The mass of the Earth is 6.0 x 1024 kg and the average distance between the Earth and the Moon is 3.8 x 108 m. 

Project 9.  SEQ Project_9. \* ARABIC 3 Communication satellites remain fixed over the same spot on the Earth’s surface. They appear to be stationary. Actually they are rotating around the Earth with a period of 24 hours. Such an orbit is called “geosynchronous” . Assuming the satellite’s orbit is circular, what is its radius of the orbit? Express your answer as a multiple of the Earth’s radius. (Use the project above. The answer is somewhere between 5 and 7 Earth radii.)

Project 9.  SEQ Project_9. \* ARABIC 4 As long ago as the third century B.C., Greek astronomers noticed that at certain times of the year the planets reverse their motion in the sky. This is called retrograde motion and is considered by astrologers to herald ominous future events. 

Retrograde motion of Mars relative to the fixed stars.

Can you construct a program which illustrates the cause of this seemingly bizarre behavior? You will need to draw two orbiting planets. Use both the turtle and the monkey. Keep the coordinates and velocities of the monkey separate from those of the turtle. 

Chapter  SEQ Chapter \* ARABIC 10

Voyager II 

Introduction


One of the great achievement of the space program was Voyager II, a space ship sent to Uranus, after Pluto the most distant of the planets in our solar system. While it took eight years to reach its destination, it would have taken much longer except for the clever device used to give the space craft a boost. 


This was acomplised by effectively “bouncing” the crft of Jupiter, using Jupiter as a paddle to hasten Voyager on its way. Now you can’t just bounce off of planet but you can use the gravitatial field of the planet to pull the craft aroud the planet as illustrated in the figure.

Jupiter acts like a large paddle, whipping Voyager around its back side and propelling it toward Uranus.

Constucting the orbit

In order to simulate this process we will again employ bother the turtle and the monkey. The turtle will simulate voyager and we will let the monkey take the place of Jupiter. 

on voyager xSatellite,ySatellite,vSatellite,angleSatellite,xPlanet,vPlanet

 global Mass

 setHome 0,0

 put 2000 into mass --Arbitrary mass chosed to keep the turtle on the screen.

 st --Ititiate turtle, a.k.a. planet.

 sm -- Initiate monkey, a.k.a. satellite.

 ct -- Clear the text.

 Pu --Pick the pen up.

 put .2 into dt

 seth angleSatellite -- These next four steps are used to get components of the salellite’s velocity.

 fd vSatellite

 put xcor() into vxSatellite

 put ycor() into vySatellite

 pr "The initial speed of voyager is " &vSatellite

 put 0 into yPlanet -- The planet moves at a steady pace along they horizontal axis.

 setxy xSatellite,ySatellite

 pd

 tellMonkey

  pu

  setxy xPlanet,yPlanet --Set the position of the planet.

 EndMonkey

 repeat until the mouseClick

  setxy xSatellite,ySatellite

  dot -- Place a dot at the position of Voyager.

  tellMonkey

   setXY xPlanet,yPlanet

   dot -- Place a dot at the position of the planet.

  endMonkey

  add vPlanet*dt to xPlanet -- Get the new position of the planet.

  add vxSatellite*dt to xSatellite -- These two lines determine the new position of Voyager.

  add vySatellite*dt to ySatellite

  add dt*accx(xPlanet - xSatellite,yPlanet-ySatellite) to vxSatellite -- Find the new velocity of Voyager. 

  add dt*accy(xPlanet - xSatellite,yPlanet-ySatellite) to vySatellite

 end repeat

 pr "The final speed of voyager is " & sqrt(vxSatellite^2+vySatellite^2)

end voyager

function accx x,y

 global Mass

 return Mass*x/(x^2+y^2)^1.5

end accx

function accy x,y

 global Mass

 return Mass*y/(x^2+y^2)^1.5

end accy

on dot

 pd

 fd 0

 pu

end dot


As a test try “voyager 130, 30 , 4, 180, 250, -12” and you should see evenly spaced dots representing Jupiter moving from right to left at constant velocity. You will also see evenly spaced dots representing Voyager coming up from the bottom left of the screen, until it comes close to Jupiter and then you will see it speed up—the dots become further apart—and be deflected off to the upper left of the screen. When you click the mouse button the final velocity of Voyager will be printed to the screen. The velocity increases by nearly a factor of 5, from 3 to 15. (These are not realistic numbers but only used to illustrate the effect.) 

Project 10.  SEQ Project_10. \* ARABIC 1 The above example shows how a space ship might pick up speed from an approaching planet. Demonstrate how a space ship would lose speed in colliding with a receding planet. Hint: Bring the space ship in from the lower right side of the screen.

Project 10.  SEQ Project_10. \* ARABIC 2 Another interesting example of getting a bump up in speed by bouncing the satellite off of a planet is the Cassini satellite launched on October 8, 1997. In this case the Cassini trajectory puts it in a path to get a boost from Venus which comes up from behind and—in effect—pushes Cassini ahead of it. Of course the force of gravity is always attractive and cannot actually push on anything. But, as shown in the figure it is possible for Venus to catch up to the satellite from behind and whip it around the back side. Experiment with the input variables to the Voyager program above to see if you can duplicate this result. (Take care not to let the satellite get too close to the planet or the numerical approximation which calculates the trajectory will break down.)
The planet comes from the right and eventually catches up with the slower satellite. The satellite is attracted to the planet and eventually falls in behind the planet where it gaterhs momentum from Venus and is propelled out ahead of the planet.

Chapter  SEQ Chapter \* ARABIC 11

Rockets, Jets, and Momentum

Introduction


If you have ever seen anyone fire a shotgun you may have noticed the kickback the shooter received from the gun. The larger the caliber of the bullet the greater the kickback. The kickback is also affected by the size of the gun. If a bullet were fired from a cannon there would be little effect on the cannon. 


If we want to study the science of kickback we need a quantitative measure, a number we can place on the magnitude of the kickback. It is not enough to say that to every action there is a reaction; to the action of the projectile’s motion there is the reaction of the gun. We need to quantify this concept of action and reaction. 
In Newton’s mind, the equality of action and reaction means that the force which the bullet exerts on the gun is equal to the force which the gun exerts on the bullet. Furthermore, since the force on any body is equal to the rate of change in momentum, it follows that the rate of increase in the momentum of the bullet is equal to the rate of decrease in the momentum of the gun. 


Now, what is momentum? Newton defined momentum as the product of the mass and the velocity.


Momentum = p = mv

Moreover, if there are no outside forces acting on a system (or the forces are relatively small and can be neglected) then the momentum must be a constant. When this is the case we have one of the most fundamental and profound principles of nature, the principle of conservation of momentum: 

In any isolated system, the total momentum remains a constant in time.

Where, by isolated, we mean there are no significant outside forces. In the case of the gun firing a bullet the only significant forces during the firing phase are the internal forces of bullet on gun and gun on bullet. (We neglect the comparatively small force of gravity acting during this period.) Therefore momentum is conserved. If the gun is originally at rest then the total momentum before firing is zero, the total momentum after firing must also be zero. The momentum the bullet after firing must be equal and opposite to the momentum of the gun. If we know the muzzle velocity of the bullet we can calculate the reaction velocity of the gun (or gun and hunter if they stay together).

Rockets


Now the title of this chapter is Jets, Rockets, and Momentum. What’s all this talk about guns and bulletes? Well, there is really no essential difference between the means by which a rocket is propelled and that by which a gun is propelled in reaction to the firing of the bullet. The rocket does not fire bullets but it does eject prpellant from its rear and that amounts to the same thing. 


To be specific, the figure shows a rocket of mass M, traveling at a speed V. It ejects a propellant of mass m with a speed v relative to the rocket. The rocket speeds up to a velocity V’. 

A Rocket
From the principle of the conservation of momentum we know that, in the absence of outside forces (we ignore gravity) the momentum of the system must be conserved, that is the initial momentum must equal the final momentum. Therefore

MV = (M - m) V’ + m (V’ -v)

Solving for V’ we find

V’ = V + mv/M

so that the increase in velocity—mv/M—is proportional to mass and velocity of the propellant, and inversely proportional to the mass of the rocket. 

Jets


Now let us contrast this immediately with a jet airplane. The jet does not carry its own propellant but instaed picks up outside air, mixes it with fuel, ignites the mixture, and the combusted mixture is propelled out the back. The bulk of the mass ejected is that of the air. Very little is fuel. The jet therefore will not function in space as does the rocket. It must pick up it propellant on the fly while the rockert is slef contained.


Now let us apply our conservation of momentum primciple to the jet to determien the rate at wich the veloocity increases. In the fireure a jet of mass M, travleing at a speed V, picks up air of mass m. The air is mixed with fuel and ejected from the rear with a vleocity v relaative to the jet. The new velocity of the jet is V’. 

A Jet
Equating momentum before and after we find:

MV = (M-m) V’ + m (V’ - v)

or, solving for V’:

V’ = V + m (v - V’) / M

so that the increase in velocity is not the same as that for the rocket. In fact we see that it can be negative, that is, the jet will actually slow down, unless the velocity of the propellant is greater than the velocity of the jet. The propellant must be faster than the jet. If it were exactly the same speed as the jet there would be no net change. Its as if there were not interaction between the jet and the air. If it is slower than the jet, the jet is slowed. This is not unlike the drag that any body experiences in moving through the air: the object causes air around it to acquire a momentum at the expense of the momentum of the body. The trick is to get the air traveling in a direction opposite to the direction of the jet.


When you think about it there are a lot of means of transport which operate on the same principle: a propeller driven plane, a motor boat, a canoe, and a row boat. In each case the propellant—air or water—is forced to move away from the vehicle at a speed greater than the vehicle. There is an upper limit on the speed of the vehicle; there is a maximum speed of the propeller, oar, or paddle. When the vehicle reaches this speed the speed of the propeller, oar or paddle relative to the water is zero and will have no further effect of the speed of the vehicle. This was the same limitation we find in the case of the jet airplane. When V’ is greater than v the increment in the speed of the jet is negative. Better if the jet shut down altogether. This limitation does not apply to the rocket however. The rocket doesn’t have to pick up it propellant; it carries its own propellant. 

Project 11. SEQ Project_11. \* ARABIC 1 Complete the rocket program below to simulate the motion of a rocket. 
on rocket v, massPropellant, massRocket

 put 0 into velRocket

 repeat

  forward velRocket

  -- Fill in this missing line.

 end repeat

end rocket

Project 11. SEQ Project_11. \* ARABIC 2 Try various values for v, massPropellant and massRocket to determine their effect on the motion of the rocket.

Compare rocket 10, 5, 1000 to rocket 10,5, 500. Can you explain why the reduction of the rocket mass has the observed effect?

Compare rocket 10,5, 1000 to rocket 10,10,1000

Compare rocket 10,5,1000 to rocket 20,5,1000

Project 11. SEQ Project_11. \* ARABIC 3 Repeat the previous two projects but for the jet rather than the rocket.

Project 11. SEQ Project_11. \* ARABIC 4 Give a jet an initial velocity which is greater than the relative velocity of the exhaust. Describe the motion.

Project 11. SEQ Project_11. \* ARABIC 5 Improve on your rocket program by including the effect of the decreasing mass of the rocket as it exhausts its fuel. 

Project 11. SEQ Project_11. \* ARABIC 6 Consider a rocket in which half the mass of a rocket is propellant. If this propellant is ejected with a velocity of 3 ft/sec relative to the rocket, what is the increase in speed of the rocket when all of the propellant is exhausted? (Notice that neither the mass of the rocket nor the mass of propellant ejected every second is given. Try different values for these unknowns and show that the answer depends only on the ratio of propellant to rocket mass.)

